## Universidad de la República, Uruguay

#### PROYECTO DE GRADO

# Desarrollo de DSLs en lenguajes con tipos dependientes

Gonzalo Waszczuk

Supervisado por Marcos VIERA Alberto PARDO

## Resumen

Pendiente: Resumen

## Índice general

| 1. | Introducción  | 1  |
|----|---|----|
|    | 1.1. Records Extensibles                              | 1  |
|    | 1.2. Tipos Dependientes                               | 2  |
|    | 1.3. Idris  | 3  |
|    | 1.4. Guía de trabajo                                  | 3  |
| 2. | Estado del arte                                       | 5  |
|    | 2.1. Records Extensibles como primitivas del lenguaje | 5  |
|    | 2.2. Records Extensibles como bibliotecas de usuario  | 7  |
|    | 2.3. Listas heterogéneas                              | 8  |
|    | 2.3.1. HList en Haskell                               | 9  |
|    | 2.3.2. HList en Idris                                 | 10 |
|    | 2.4. Records Extensibles en Haskell                   | 11 |
| 3. | Records Extensibles en Idris                          | 14 |
|    | 3.1. HList actualizado                                | 14 |
|    | 3.1.1. Tipos Decidibles                               | 15 |
|    | 3.2. Definicion de Records Extensibles                | 15 |
|    | 3.3. Extension de records dinámicamente               | 17 |
|    | 3.3.1. Construccion de terminos de prueba             | 17 |
|    | 3.3.2. Generacion de pruebas automática               | 19 |
|    | 3.4. Proyeccion de un record                          | 22 |
|    | 3.4.1. Computacion de funcion en el tipo del record   | 30 |
|    | 3.5. Busqueda de un elemento en un record             | 32 |
| 4. | Caso de estudio                                       | 36 |
|    | 4.1. Introducción                                     | 36 |
|    | 4.2. Definición de una expresión                      | 36 |
|    | 4.2.1. Construcción de una expresión                  | 37 |
|    | 4.3. Interpretación de una expresión                  | 39 |
| 5. | Conclusión  | 40 |
|    | 5.1. Alternativas de HList en Idris                   | 40 |
|    | 5.1.1. Dinámico                                       | 40 |
|    | 5.1.2. Existenciales                                  | 41 |
|    | 5.1.3. Estructurado                                   | 41 |
| 6. | Trabaio a futuro                                      | 42 |

| ÍNDICE GENERAL                      |    | iii             |
|-------------------------------------|----|-----------------|
| <b>7. Apéndice</b> 7.1. Código fuen | te | <b>14</b><br>44 |

## Capítulo 1

## Introducción

Los records extensibles son una herramienta muy útil en la programación. Surgen como una respuesta a un problema que tienen los lenguajes de programacion estáticos en cuanto a records: ¿Cómo puedo modificar la estructura de un record ya definido?

En los lenguajes de programación fuertemente tipados modernos no existe una forma primitiva de definir y manipular records extensibles. Algunos lenguajes ni siquiera permiten definirlos.

Este trabajo se enfoca en presentar una manera de definir records extensibles en lenguajes funcionales fuertemente tipados. En particular, se presenta una manera de hacerlo utilizando un lenguaje de programación con *tipos dependientes* llamado ldris

#### 1.1. Records Extensibles

En varios lenguajes de programación con tipos estáticos, es posible definir una estructura estática llamada 'record'. Un record es una estructura heterogénea que permite agrupar valores de varios tipos en un único objeto, asociando una etiqueta o nombre a cada uno de esos valores. Un ejemplo de la definición de un tipo record en Haskell sería la siguiente:

```
data Persona = Persona { edad :: Int, nombre :: String}
```

En muchos lenguajes de programación, una desventaja que tienen los records es que una vez definidos, no es posible modificar su estructura de forma dinámica. Tomando el ejemplo anterior, si uno quisiera tener un nuevo record con los mismos campos de Persona pero con un nuevo campo adicional, como apellido, entonces uno solo podría definirlo de una de estas dos formas:

Definir un nuevo record con esos 3 campos.

```
data Persona2 = Persona2 {
  edad :: Int,
  nombre :: String,
  apellido: String
}
```

■ Crear un nuevo record que contenga el campo apellido y otro campo de tipo Persona.

```
data Persona2 = Persona2 {
  datosViejos: Persona,
  apellido : String
}
```

En ninguno de ambos enfoques es posible obtener el nuevo record extendiendo el anterior de manera dinámica. Es decir, siempre es necesario definir un nuevo tipo, indicando que es un record e indicando sus campos.

Los records extensibles intentan resolver este problema. Si Persona fuera un record extensible, entonces definir un nuevo record idéntico a él, pero con un campo adicional, sería tan fácil como tomar un record de tipo Persona existente, y simplemente agregarle el nuevo campo apellido con su valor correspondiente. A continuación se muestra un ejemplo de record extensible, con una sintaxis hipotética similar a Haskell:

```
p :: Persona
p = Persona { edad = 20, nombre = "Juan" }

pExtensible :: Persona + { apellido :: String }
pExtensible = p + { apellido = "Sanchez" }
```

El tipo del nuevo record debería reflejar el hecho de que es una copia de Persona pero con el campo adicional utilizado. Uno debería poder, por ejemplo, acceder al campo apellido del nuevo record como uno lo haría con cualquier otro record arbitrario.

Para poder implementar un record extensible, es necesario poder codificar toda la información del record. Esta información incluye las etiquetas que acepta el record, y el tipo de los valores asociados a cada una de esas etiquetas.

Algunos lenguajes permiten implementar records extensibles con primitivas del lenguaje (como en Elm [14] o Purescript [23]), pero requieren de sintaxis y semántica especial para records extensibles, y no es posible implementarlos con primitivas del lenguaje más básicas.

Las propiedades de records extensibles requieren que los campos sean variables y puedan ser agregados a records previamente definidos. En el ejemplo visto más arriba, básicamente se espera poder tener un tipo Persona, y poder agregarle un campo apellido con el tipo String, y que esto sea un nuevo tipo. Esto se puede realizar con tipos dependientes.

## 1.2. Tipos Dependientes

Tipos dependientes son tipos que dependen de valores, y no solamente de otros tipos. En el ejemplo anterior, el nuevo tipo Persona + { apellido :: String} depende de tipos (Persona y String) pero también de valores. El elemento apellido se puede representar como un valor de tipo String, quedando el tipo final Persona + { 'apellido' :: String }. Con tipos dependientes los tipos son first-class, lo que permite que cualquier operación que se puede

realizar sobre valores también se puede realizar sobre tipos. En particular, se pueden definir funciones que tengan tipos como parámetros y retornen tipos como resultado. En el ejemplo, + funciona como tal función, tomando el tipo Persona y el tipo {'apellido':: String} y uniéndolos en un nuevo tipo.:: funciona como otra función, uniendo el valor apellido y el tipo String.

En relacion a records extensibles, los tipos dependientes hacen posible que dentro del tipo del record puedan existir valores para poder ser manipulados, como podrian ser la lista de etiquetas del record. Como esta informacion se encuentra en el tipo del record, es posible definir tipos y funciones que accedan a ella y la manipulen para poder definir todas las funcionalidades de records extensibles.

Otra propiedad de los tipos dependientes es que permiten definir propiedades como tipos. Es posible definir proposiciones como tipos, y probar tales proposiciones construyendo valores de ese tipo. Como ejemplo, se puede codificar la propiedad 'Un record extensible no puede tener campos repetidos' como un tipo RecordSinRepetidos record (donde record es un valor de tipo record), y luego se puede usar un valor prueba: RecordSinRepetidos record en cualquier lugar donde uno quiera que se cumpla esa propiedad. Como se verá más adelante en el trabajo, esto permite tener chequeos y restricciones sobre la construcción de records en tiempo de compilación, ya que si un tipo que representa una propiedad no puede ser construido (lo que significa que no se pudo probar esa propiedad), entonces el código no va a compilar. Esto es una mejora a la alternativa de que se realice un chequeo en tiempo de ejecución haciendo que falle el programa, ya que uno no necesita correr el programa para saber que éste es correcto y cumple tal propiedad.

#### 1.3. Idris

En este trabajo se decidio utilizar el lenguaje de programacion *Idris* [10] para llevar a cabo la investigacion del uso de tipos dependientes aplicados a la definicion de records extensibles. Idris es un lenguaje de programacion funcional con tipos dependientes, con sintaxis similar a Haskell.

Otro lenguaje que cumple con los requisitos es *Agda* [22], un lenguaje funcional con tipos dependientes. Sin embargo, se decidió utilizar Idris para investigar las funcionalidades de este nuevo lenguaje. Por este motivo, todas las conclusiones obtenidas en este informe pueden ser aplicadas a *Agda* también.

La version de Idris utilizada en este trabajo es la 0.12.0.

#### 1.4. Guía de trabajo

Este trabajo se divide en varias secciones.

En el capítulo *Estado del arte* se van a mostrar varias implementaciones de records extensibles en otros lenguajes, mostrando sus beneficios e inconvenientes. Eventualmente se enfocará en una implementación específica de records extensibles en Haskell.

En el capítulo *Records Extensibles en Idris* se presentará la implementación de records extensibles en Idris llevada a cabo en este proyecto, expandiendo en el uso de Idris y las funcionalidades de éste que hicieron posible este trabajo, y mostrando los problemas encontrados y sus respectivas soluciones.

En el capítulo *Caso de Estudio* se presentará un caso de estudio de construcción y evaluación de expresiones aritméticas, haciendo uso de lo desarrollado en este trabajo, permitiendo ver cómo es el uso de esta implementación y qué se puede llegar a hacer con ella.

En el capítulo *Conclusión* se tomarán las experiencias del desarrollo del trabajo y el caso de estudio, y se analizarán los problemas encontrados, el aporte que puede llegar a tener este trabajo sobre el problema de records extensibles, etc.

En el capítulo *Trabajo a futuro* se indicarán aspectos de diseño y implementación que pueden realizarse sobre este trabajo para mejorarlo, como posibles tareas que surgieron y se pueden llevar a cabo.

En el *Apéndice* se encontrarán recursos adicionales, como el código fuente de este trabajo.

## Capítulo 2

## Estado del arte

En este capítulo se van a presentar varias implementaciones de records extensibles en varios lenguajes. Se enfocará en lenguajes fuertemente tipados. Existen implementaciones de records extensibles en lenguajes dinámicamente tipados, pero no entran en el alcance de este trabajo ni de lo que se quiere investigar.

En términos generales, las implementaciones de records extensibles se dividen según si son proporcionados por el lenguaje como primitivas, o si son proporcionados por bibliotecas. Se describirán ambas alternativas, incluyendo ejemplos de lenguajes e implementaciones de cada uno.

En particular, este trabajo se enmarca dentro de los proporcionados como bibliotecas de usuario. Este trabajo fue motivado por una biblioteca de Haskell llamada *HList*, la cual va a ser presentada más adelante.

## 2.1. Records Extensibles como primitivas del lenguaje

Algunos lenguaje de programación funcionales permiten el manejo de records extensibles como primitiva del lenguaje. Esto significa que records extensibles son funcionalidades del lenguaje en sí, y tienen sintaxis y funcionamiento especial en el lenguaje.

Uno de ellos es Elm[12]. Uno de los ejemplos que se muestra en su documentación ([14]) es el siguiente:

```
type alias Positioned a =
  { a | x : Float, y : Float }

type alias Named a =
  { a | name : String }

type alias Moving a =
  { a | velocity : Float, angle : Float }

lady : Named { age:Int }

lady =
```

Elm permite definir tipos que equivalen a records, pero agregándole campos adicionales, como es el caso de los tipos descritos arriba. Este tipo de record extensible hace uso de *row polymorphism*. Básicamente, al definir un record, éste se hace polimórfico sobre el resto de los campos (o *rows*, como es descrito en la literatura). Es decir, se puede definir un record que traiga como mínimo unos determinados campos, pero el resto de éstos puede variar. En el ejemplo de arriba, el record lady es definido extendiendo Named con otro campo adicional, sin necesidad de definir un tipo nuevo.

Una desventaja es que por el poco uso de records extensibles en la versión 0.16 se decidió eliminar la funcionalidad de agregar y eliminar campos a records de forma dinámica [13].

Otro lenguaje con esta particularidad es *Purescript* [24]. A continuación se muestra uno de los ejemplos de su documentación [23]:

```
fullname :: forall t. { firstName :: String,
   lastName :: String | t } -> String
fullName person = person.firstName ++ " " ++ person.lastName
```

Purescript permite definir records con determinados campos, y luego definir funciones que solo actúan sobre los campos necesarios. Utiliza *row polymorphism* al igual que Elm.

Ambos lenguajes basan su implementación de records extensibles, aunque sea parcialmente, en el paper Extensible records with scoped labels [20], de Daan Leijen.

También existen otras propuestas de sistemas de tipos con soporte para records extensibles. En *First-class labels for extensible rows*[21], Daan Leijen describe un sistema de tipos con etiquetas como *first-class citizens* del lenguaje para permitir construir records extensibles de forma expresiva. En *A Polymorphic Type System for Extensible Records and Variants* [15], Benedict R. Gaster y Mark P. Jones también describen un sistema de tipos con soporte para records extensibles. Ambos se basan en extensiones del sistema de tipos de Haskell y ML, incluyendo inferencia de tipos. Este paper de Gaster y Jones fue utilizado para la extensión Trex del compilador Hugs98 [9]. Hugs98 es un compilador de Haskell, cuya extensión Trex soporta records extensibles utilizando los fundamentos teóricos del paper de Gaster y Jones. Desafortunadamente, Hugs98 no está más en desarrollo. Otra propuesta es la de *Operations on Records*[11], de Luca Cardelli y John C. Mitchell. Esta propuesta crea una teoría de records extensibles, formalizando un sistema de tipos para soportarlos. Otra propuesta es *Lightweight Extensible Records for Haskell*[17], de Mark P. Jones

y Simon Peyton Jones, la cual describe una extensión de Haskell98 para soportar records extensibles. Muchas de las propuestas son similares y utilizan los mismos fundamentos de la teoría de records.

Estas propuestas, al igual que cualquier propuesta de extender un lenguaje para que soporte extensible records como primitivas, tienen algunas desventajas. La desventaja principal es que se deben agregar nuevas reglas de tipado para soportar los records, y esto puede impactar otros aspectos del lenguaje. En el paper Lightweight Extensible Records for Haskell[17], los autores describen un problema de su extensión de records, donde al tener un nuevo kind (tipo de tipos) sobre records, existen ambiguedades y problemas al integrarlo con el sistema de typeclasses de Haskell. Al agregar una nueva regla al lenguaje para soportar records, esta regla puede llegar a ser inconsistente con otras partes del lenguaje, y puede requirir un rediseño de muchas partes del lenguaje para que funcionen correctamente con estas nuevas reglas. Esto también genera que el lenguaje sea más costoso de mantener y entender, lo cual puede traer problemas futuros cuando se intenten implementar nuevas funcionalidades y extensiones del lenguaje. Otro problema es que, al menos que se explicite, las nuevas funcionalidades del lenguaje que soportan records extensibles no son first-class. Esta falta de soporte limita la expresividad de tales records, ya que cualquier manipulación de ellos debe ser proporcionada por el lenguaje, cuando podrían ser proporcionadas por el usuario mismo.

Este trabajo se enfocará en records extensibles como bibliotecas de usuario. Una biblioteca no modifica el lenguaje mismo sino que hace uso de todas sus funcionalidades ya definidas, por lo que no presentan los problemas vistos anteriormente. A su vez las bibliotecas le dan varias opciones al usuario, que puede elegir la implementación de records extensibles que desea y le resulta mejor para su situación. Otra ventaja es que poder definir records extensibles como una biblioteca evita que uno necesite cambiar el lenguaje para poder soportarlos.

### 2.2. Records Extensibles como bibliotecas de usuario

Las bibliotecas de usuario son componentes de un lenguaje de programación escrito en ese lenguaje para proveer una funcionalidad al usuario. Las implementaciones de records extensibles que utilizan este mecanismo se basan en utilizar funcionalidades avanzadas del lenguaje en cuestión para poder realizar la definición de tales records.

En Generic Record Combinators with Static Type Checking [16], Wolfgang Jeltsch describe un sistema de records en Haskell definiendo los records como typeclasses, y los tipos y etiquetas del record como tipos producto. Su trabajo fue liberado como una biblioteca de Haskell llamada records [1]. Esta implementación es similar a HList, pero no nos enfocaremos en ella en este trabajo.

Otras propuestas se basan en utilizar extensiones de Haskell para poder definir los records. rawr [2] utiliza type families y type-level lists para definir sus records. Vinyl [3] también utiliza type-level lists, pero define el record como un GADT (Generalized Abstract Data Type). ruin [4] utiliza Template Haskell y metaprogramming para crear instancias de type-classes específicas. Existen otras propuestas, como labels [5], named-records [6], bookkeeper [7], entre otras. Estas bibliotecas son muy similares en su funcionamiento, difiriendo en qué funcionalidades de GHC (Glasgow Haskell

Compiler) y Haskell utilizan, qué funcionalidades de records y records extensibles le proporcionan al usuario, qué sintaxis usan, etc.

Como ejemplo de uso de una biblioteca, se tienen los records de *rawr* que se definen de la siguiente forma:

```
type Foo = R ( "a" := Int, "b" := Bool )
let foo = R ( #a := 42, #b := True ) :: Foo
```

La definición de un record se realiza llamando a funciones y tipos proporcionados por la biblioteca (como es el caso de las funciones  $\mathbb R$  en este ejemplo). Para realizar la extensión de un record, generalmente también se utilizan operadores definidos en la biblioteca, como en el siguiente caso:

```
R ( #foo := True ) :*: R ( #bar := False )
--> R ( bar := False, foo := True )
```

Existen muchas alternativas y propuestas, pero el trabajo actual se basó en la biblioteca HList de Haskell. HList muestra las problemáticas de definir records extensibles, a la vez que muestra soluciones a ellas. También tiene varias propiedades que son deseadas tener en records extensibles (como flexibilidad para agregar nuevas operaciones sobre records), y el marco de este trabajo es transladarlo a un contexto de tipo dependientes en Idris y hacer uso de esas propiedades.

Uno de los mecanismos utilizados por HList y la mayoría de las bibliotecas previamente descritas son las listas heterogéneas, que se presentarán más adelante. Bibliotecas como HList y Records (de Wolfgang Jeltsch) son más viejas y utilizan funcionalidades más básicas de Haskell para definir listas heterogéneas (como type families y type classes). Algunas de las bibliotecas mencionadas (como vinyl) son más modernas y utilizan funcionalidades nuevas de GHC, como DataKinds, que permite definir listas heterogéneas a nivel de tipos, y así permitir definir records extensibles de forma más sencilla.

Las listas heterogéneas son un pilar base de muchas implementaciones de records extensibles, pero en particular son fundamentales para la implementación de HList, y subsecuentemente para la implementación de records extensibles en Idris. A continuación se describirán las listas heterogéneas, luego se mostrará cómo se implementan en la biblioteca HList de Haskell, y luego se presentará cómo son implementadas en Idris.

## 2.3. Listas heterogéneas

El concepto de listas heterogéneas (o *HList*) surge en oposición al tipo de listas más utilizado en la programación con lenguajes tipados: listas homogéneas. Las listas homogéneas son las más comunes de utilizar en estos lenguajes, ya que son listas que pueden contener elementos de un solo tipo. Estos tipos de listas existen en todos o casi todos los lenguajes de programación que aceptan tipos parametrizables o genéricos, sea Java, C#, Haskell y otros. Ejemplos comunes son List<int> en Java, o [String] en Haskell.

Las listas heterogéneas, sin embargo, permiten almacenar elementos de cualquier tipo arbitrario. Estos tipos pueden o no tener una relación entre ellos, aunque en

general se llama 'listas heterogénea' a la estructura de datos que no impone ninguna relación entre los tipos de sus elementos.

En lenguajes dinámicamente tipados (como lenguajes basados en LISP) este tipo de listas es fácil de construir, ya que no hay una imposición por parte del interprete sobre qué tipo de elementos se pueden insertar o pueden existir en esta lista. El siguiente es un ejemplo de lista heterogénea en un lenguaje LISP como Clojure o Scheme, donde a la lista se puede agregar un entero, un float y un texto.

```
'(1 0.2 "Text")
```

Sin embargo, en lenguajes tipados este tipo de lista es más dificil de construir. Para determinados lenguajes no es posible incluir la mayor cantidad de información posible en tales listas para poder trabajar con sus elementos, ya que sus sistemas de tipos no lo permiten. Esto se debe a que tales listas pueden tener elementos con tipos arbitrarios, por lo tanto los sistemas de tipos de estos lenguajes necesitan una forma de manejar tal conjunto arbitrario de tipos.

En sistemas de tipos más avanzados, es posible incluir la mayor cantidad posible de información de tales tipos arbitrarios en el mismo tipo de la lista heterogénea.

A continuación se describe la forma de crear listas heterogéneas en Haskell, y luego la forma de crearlas en Idris, utilizando el poder de tipos dependientes de este lenguaje para llevarlo a cabo.

#### 2.3.1. HList en Haskell

Para definir listas heterogéneas en Haskell nos basaremos en la propuesta presentada en *Strongly Typed Heterogeneous Collections* [19], por Oleg Kiselyov, Ralf Lämmel y Keean Schupke. Esta propuesta está implementada en el paquete *hlist*, encontrado en Hackage [18].

Esta biblioteca define HList de la siguiente forma:

```
data family HList (l::[*])

data instance HList '[] = HNil
data instance HList (x ': xs) = x `HCons` HList xs
```

Se utilizan data families para poder crear una familia de tipos HList, utilizando la estructura recursiva de las listas, definiendo un caso cuando una lista está vacía y otro caso cuando se quiere agregar un elemento a su cabeza. Esta definición representa una secuencia de tipos separados por HCons y terminados por HNil.

Esto permite, por ejemplo, construir el siguiente valor:

```
10 `HCons` ('Text' `HCons` HNil) :: HList '[Int, String]
```

Esta definición hace uso de una forma de tipos dependientes, al utilizar una lista de tipos como parámetro de HList. La estructura recursiva de HList garantiza que es posible construir elementos de este tipo a la misma vez que se van construyendo elementos de la lista de tipos que se encuentra parametrizada en HList.

Para poder definir funciones sobre este tipo de listas, es necesario utilizar *type* families, como muestra el siguiente ejemplo

```
type family HLength (x :: [k]) :: HNat
type instance HLength '[] = HZero
type instance HLength (x ': xs) = HSucc (HLength xs)
```

Type families como la anterior permiten que se tenga un tipo <code>HLength ls</code> en la definicion de una funcion y el typechecker decida cual de las instancias anteriores debe llamar, culminando en un valor del kind <code>HNat</code>.

Un ejemplo de uso de tal función sería el siguiente:

```
HLength '[Int, String] = HSucc (HSucc HZero)
```

#### 2.3.2. HList en Idris

Uno de los objetivos de la investigación de listas heterogéneas en un lenguaje como Idris es poder utilizar el poder de su sistema de tipos. Esto evita que uno tenga que recurrir al tipo de construcciones de Haskell (type families, etc) para poder manejar listas heterogéneas. Una de las metas es poder programar utilizando listas heterogéneas con la misma facilidad que uno programa con listas homogéneas. Esto último no ocurre en el caso de HList en Haskell, ya que el tipo de HList es definido con data families, y funciones sobre HList son definidas con type families. Esta definición se contrasta con las listas homogéneas, cuyo tipo se define como un tipo común, y funciones sobre tales se definen como funciones normales también.

A diferencia de Haskell, el lenguaje de programación Idris maneja tipos dependientes completos. Esto significa que cualquier tipo puede estar parametrizado por un valor, y este tipo es un first-class citizen que puede ser utilizado como cualquier otro tipo del lenguaje. Esto significa que para Idris no hay diferencia en el trato de un tipo simple como String y en el de un tipo complejo con tipos dependientes como Vect 2 Nat (este tipo se verá a continuación).

Para conocer mejor el uso de tipos dependientes en Idris, veamos la definición del tipo *vector* que son listas cuyo largo es anotado en su tipo:

```
data Vect : Nat -> Type -> Type where
[] : Vect 0 A
  (::) : (x : A) -> (xs : Vect n A) -> Vect (n + 1) A
```

Un valor del tipo Vect 1 String es una lista que contiene 1 string, mientras que un valor del tipo Vect 10 String es una lista que contiene 10 strings. En la definición Vect : Nat ->Type ->Type el tipo mismo queda parametrizado por un natural además de un tipo cualquiera.

Tener valores en el tipo permite poder restringir el uso de funciones a determinados tipos que tengan valores específicos. Como ejemplo se tiene la siguiente funcion:

```
head: Vect (n + 1) a -> a head (x :: xs) = x
```

Esta funcion obtiene el primer valor de un vector. Es una funcion total, ya que restringe su uso solamente a vectores que tengan un largo mayor a 0, por lo que tal vector siempre va a tener por lo menos un elemento para obtener. Si se intenta llamar a esta función con un vector sin elementos, como head [], la llamada no

va a compilar porque el type-checker no va a saber unificar  $Vect\ 0$  a con  $Vect\ (n+1)$  a (no puede encontrar un valor natural n que cumpla n+1=0, por lo que falla el type-checking).

En cuanto a listas heterogéneas, la definición de HList utilizada en Idris es la siguiente:

```
data HList : List Type -> Type where
  Nil : HList []
  (::) : t -> HList ts -> HList (t :: ts)
```

Esta definición permite construir listas heterogéneas con relativa facilidad, como por ejemplo:

```
23 :: "Hello World" :: [1,2,3] : HList [Nat, String, [Nat]]
```

El tipo HList se define como una función de tipo que toma una lista de tipos (e.j [Nat, String]) y retorna un tipo. Éste se construye definiendo una lista vacía que no tiene tipos, o definiendo un operador de *cons* que tome un valor, una lista previa, y agregue ese valor a la lista. En el caso de *cons*, no solo agrega el valor a la lista, sino que agrega el tipo de tal valor a la lista de tipos que mantiene HList en su tipo.

Cada valor agregado a la lista tiene un tipo asociado que es almacenado en la lista del tipo. 23 : Nat guarda 23 en la lista pero Nat en el tipo, de forma que uno siempre puede recuperar ya sea el tipo o el valor si se quieren utilizar luego.

A su vez, a diferencia de Haskell, la posición de *first-class citizens* de los tipos dependientes en Idris permite definir funciones con pattern matching sobre HList.

```
hLength : HList ls \rightarrow Nat
hLength Nil = 0
hLength (x :: xs) = 1 + (hLength xs)
```

Aquí surge una diferencia muy importante con la implementación de HList de Haskell, ya que en Idris hLength puede ser utilizada como cualquier otra función, y en especial opera sobre *valores*, mientras que en Haskell HLength solo puede ser utilizada como type family, y solo opera sobre *tipos*. Esto permite que el uso de HList en Idris no sea distinto del uso de listas comunes (List), y por lo tanto puedan tener el mismo trato de ellas para los creadores de bibliotecas y los usuarios de tales, disminuyendo la dificultad de su uso para resolver problemas.

A continuación se presentará cómo la biblioteca de Haskell HList define records extensibles utilizando estas listas heterogéneas.

#### 2.4. Records Extensibles en Haskell

La propuesta de HList [19] utiliza listas heterogéneas para definir un record:

```
newtype Record (r :: [*]) = Record (HList r)
mkRecord :: HRLabelSet r => HList r -> Record r
```

```
mkRecord = Record
```

Un record se representa simplemente como una lista heterogénea de un tipo determinado (que se verá más adelante). Un record se puede construir solamente utilizando mkRecord. Esta función toma una lista heterogénea, pero fuerza a que tenga una instancia de HRLabelSet

Una lista heterogénea con una instancia para esta type-class implica que la lista tiene valores con etiquetas, y ninguna etiqueta se repite. Se define de la siguiente forma:

```
class (HLabelSet (LabelsOf ps), HAllTaggedLV ps) =>
   HRLabelSet (ps :: [*])
```

Para poder implementar una instancia de esta type-class, la lista necesita cumplir el predicado HAllTaggedLV y HLabelSet. Para esto la lista debe contener este tipo:

```
data Tagged s b = Tagged b
```

Este tipo permite tener un *phantom type* en el tipo s. Esto significa que un valor de tipo Tagged s b va a contener solamente un valor de tipo b, pero en tiempo de compilación se va a tener el tipo s para manipular.

El predicado HAllTaggedLV simplemente verifica que la lista solo contenga elementos del tipo Tagged. Ambos Tagged y HAllTaggedLV pertenecen a la biblioteca tagged [8].

Un ejemplo posible de tal lista sería

```
Tagged 10 `HCons` Tagged 'John' `HCons` HNil ::
   HList '[Tagged s1 Nat, Tagged s2 String]
```

El valor en sí no contiene las etiquetas s1 ni s2, pero el tipo las contiene. Estas etiquetas no deben repetirse, y para eso la lista debe poder tener una instancia de HLabelSet (LabelsOf ps). LabelsOf es una type-family que toma una lista de Tagged y obtiene sus etiquetas:

```
type family LabelsOf (ls :: [*]) :: [*]
type instance LabelsOf '[] = '[]
type instance LabelsOf (Label l ': r) = Label l ': LabelsOf r
type instance LabelsOf (Tagged l v ': r) = Label l ': LabelsOf r
```

LabelsOf ls toma todos los phantom type de Tagged y los retorna en una lista a nivel de tipos. Como en algunas partes de la implementación se permite que la lista heterogénea tenga solo etiquetas (sin valores) tal lista puede tener un valor del tipo Label l. Este tipo Label simplemente permite construir una etiqueta a nivel de tipos para poder identificar los campos del record.

HLabelSet es una type-class que representa el predicado de que las etiquetas de la lista no estén repetidas. Para ello se aplica la función LabelsOf antes, para poder obtener solo las etiquetas de la lista (y no sus valores). Este predicado se define de forma recursiva definiendo instancias para cada caso base y caso inductivo. Hace uso de predicados de igualdad de tipos para poder realizarlo. No se mostrará su implementación.

La biblioteca también proporciona otra forma de generar records utilizando etiquetas y operadores especiales. Un ejemplo de su uso es:

```
data PersonaNamespace = PersonaNamespace
clave = firstLabel PersonaNamespace "clave"
nombre = nextLabel clave "nombre"
edad = nextLabel nombre "edad"

persona = clave .=. (3 :: Integer)
    .*. nombre .=. 'Juan'
    .*. edad .=. 27
    .*. emptyRecord
```

Funciones como firstLabel y nextLabel permiten construir elementos del tipo Label t, que luego pueden usarse para definir los campos del record con el siguiente operador:

```
(.=.) :: Label 1 -> v -> Tagged 1 v
```

El operador (.\*.) permite extender un record con un nuevo campo. Su implementación no se mostrará, pero cabe notar que su implementación termina realizando un llamado a la función mkRecord definida anteriormente.

Todas las funciones sobre records extensibles se definen utilizando typeclasses para realizar la computación en los tipos. Un ejemplo de ello es la función que obtiene un elemento de un record dada su etiqueta:

```
class HasField (l::k) r v | l r -> v where
  hLookupByLabel:: Label l -> r -> v

(.!.) :: (HasField l r v) => r -> Label l -> v
r .!. l = hLookupByLabel l r
```

Para este trabajo se tomó como motivación esta implementación de records, realizando una traducción a Idris de cada implementación, cada tipo y cada algoritmo. A continuación se mostrará la implementación realizada en este trabajo.

## Capítulo 3

## Records Extensibles en Idris

#### 3.1. HList actualizado

En este trabajo se decidio implementar las listas heterogeneas de una forma distinta a la que utiliza Idris en general. Las listas heterogeneas de Idris permiten incluir cualquier tipo arbitrario, pero eso no es suficiente al momento de poder implementar records extensibles. Para implementar records extensibles, no solo es necesario poder tener tipos arbitrarios en tal lista, sino que es necesario asociar una etiqueta a cada uno de esos tipos.

Una posible implementacion es la de Record de Haskell definida en el capítulo anterior. En esta se define un tipo Tagged s b que contenga un phantom type, de tal forma que luego se utiliza una lista de esos tipos. Sin embargo se decidio no utilizar esta implementacion, ya que necesita definir tipos auxiliares (como Tagged), y a su vez necesita definir predicados específicos para esta implementacion, como el predicado que indica que la lista heterogenea está compuesta solamente por valores del tipo Tagged.

Se decidio utilizar una solucion más simple, en donde se reimplementa <code>HList</code> para que contenga la etiqueta junto al tipo

El tipo LabelList representa una lista de tipos, junto a una etiqueta para cada uno de esos tipos. Las etiquetas pueden ser de cualquier tipo, pero generalmente se utiliza String para poder nombrarlas mejor. Un ejemplo de tal lista sería [("1", Nat),("2", String)]: LabelList String. El resto de la implementacion es identica a la de HList de Idris, con la diferencia de que se debe pasar no solo el valor sino la etiqueta de su campo tambien.

En Idris se pueden definir parametros implicitos, como  $\{1: lty\}$ , a diferencia de parametros explicitos como (val:t). Al momento de utilizar un tipo o una funcion no es necesario incluir los parametros implicitos, Idris va a intentar calcularlos

automaticamente basados en el contexto de la llamada o uso del tipo o funcion. Si ldris no puede calcularlos, el typechecker va a fallar y va a ser necesario explicitar tales parametros de esta forma: {parametro=valor}.

#### 3.1.1. Tipos Decidibles

Para poder definir records extensibles en Idris, es necesario trabajar con tipos o proposiciones decidibles. En Idris la decidibilidad de un tipo se representa con el siguiente predicado:

```
data Dec : Type -> Type where
  Yes : (prf : prop) -> Dec prop
  No : (contra : Not prop) -> Dec prop
```

Un tipo es decidible si se puede construir un valor de si mismo, o se puede construir un valor de su contradicción. Si se tiene  $Dec\ P$ , entonces significa que o bien existe un valor s: P o existe un valor  $n: Not\ P$ . Cabe notar que  $Not\ P$  es equivalente a P ->Void, representando la contradicción de P.

Poder obtener tipos decidibles es importante cuando se tienen tipos que funcionan como predicados y se necesita saber si ese predicado se cumple o no. Solo basta tener  $\texttt{Dec}\ \texttt{P}$  para poder realizar un análisis de casos, uno cuando P es verdadero y otro cuando no.

Otra funcionalidad importante es la de poder realizar igualdad de valores:

```
interface DecEq t where
  total decEq : (x1 : t) -> (x2 : t) -> Dec (x1 = x2)
```

 $\mathtt{DecEq}$  t indica que, siempre que se tienen dos elementos  $\mathtt{x1}$ ,  $\mathtt{x2}$ : t, siempre es posible tener una prueba de que son iguales o una prueba de que son distintos. La función  $\mathtt{decEq}$  es importante cuando se quiere realizar un análisis de casos sobre la igualdad de dos elementos, un caso donde son iguales y otro caso donde se tiene una prueba de que no lo son.

Los tipos decidibles y las funciones que permiten obtener valores del estilo Dec P son muy importantes al momento de probar teoremas y manipular predicados.

#### 3.2. Definicion de Records Extensibles

Al igual que la implementacion de records extensibles en Haskell vista anteriormente, en este trabajo se decidio implementar records extensibles utilizando una lista heterogenea, cuyo tipo contiene una lista de etiquetas que no deben ser repetidas.

Para comenzar con esta implementacion, se necesita que las etiquetas de los records puedan ser comparadas para verificar que no sean repetidas. En este trabajo se decidio utilizar la *typeclass* <code>DecEq lty</code> para las etiquetas de tipo <code>lty</code>. Como se vió en la sección anterior, al ser las etiquetas de un tipo que tienen una instancia de esa typeclass, la igualdad de cualquier par de valores de etiquetas es decidible, por lo que las etiquetas pueden ser comparadas.

Como en la implementacion en Haskell se decide utilizar un predicado que indica que una lista de valores no tiene repetidos. En Idris es posible codificar esta propiedad en un tipo de datos:

```
data IsSet : List t -> Type where
  IsSetNil : IsSet []
  IsSetCons : Not (Elem x xs) -> IsSet xs ->
        IsSet (x :: xs)
```

El tipo IsSet ls funciona efectivamente como un predicado logico, el cual indica que la lista ls es un conjunto que no tiene elementos repetidos.

Las pruebas de este predicado se construyen de forma constructiva. Primero se prueba que la lista vacía no contiene repetidos, y luego para el caso recursivo si se agrega un elemento a una lista, la lista resultante no va a tener repetidos solamente si el elemento a agregar no se encuentra en la lista original.

Como para el caso de records extensibles se manejan listas de pares de etiquetas y tipos, se definen los siguientes tipos y funciones útiles para manejarlos:

```
ElemLabel : lty -> LabelList lty -> Type
ElemLabel l ts = Elem l (labelsOf ts)

isElemLabel : DecEq lty => (l : lty) ->
   (ts : LabelList lty) ->
   Dec (Elem l (labelsOf ts))
isElemLabel l ts = isElem l (labelsOf ts)

labelsOf : LabelList lty -> List lty
labelsOf = map fst

IsLabelSet : LabelList lty -> Type
IsLabelSet ts = IsSet (labelsOf ts)
```

El predicado IslabelSet ts indica que para la lista de etiquetas y tipos ts no existen etiquetas repetidas. El predicado ElemLabel l ts indica que la etiqueta l pertenece a la lista de etiquetas ts. isElemLabel es una funcion de decision, donde para cualquier etiqueta y cualquier lista, se puede probar que tal etiqueta pertenece a esa lista o no.

Obtener una definicion de record extensible es simplemente necesitar una lista heterogenea etiquetada, y una prueba de que las etiquetas no son repetidas.

```
data Record : LabelList lty -> Type where
    MkRecord : IsLabelSet ts -> HList ts -> Record ts
Un caso base que se puede definir es el record vacío

emptyRec : Record []
emptyRec = MkRecord IsSetNil {ts=[]} []
```

A su vez se pueden crear funciones que proyectan sobre los campos del constructor del record. En particular, se puede convertir un record a un HList, y dado un record se puede obtener la prueba de que sus etiquetas forman un conjunto.

```
recToHList : Record ts -> HList ts
recToHList (MkRecord _ hs) = hs
```

```
recLblIsSet : Record ts -> IsLabelSet ts
recLblIsSet (MkRecord lsIsSet _ ) = lsIsSet
```

#### 3.3. Extension de records dinámicamente

La funcionalidad más importante de records extensibles es la de poder extender tal record con un nuevo campo. Esta funcionalidad puede definirse de esta forma:

```
consRec : DecEq lty => {ts : LabelList lty} ->
  {t : Type} -> (l : lty) -> (val : t) ->
  Record ts -> {notElem : Not (ElemLabel l ts)} ->
  Record ((l,t) :: ts)
consRec l val (MkRecord subLabelSet hs) {notElem} =
  MkRecord (IsSetCons notElem subLabelSet) (val :: hs)
```

Antes que nada se necesita que el tipo de etiquetas admita igualdad, utilizando la restriccion DecEq lty. Luego, se necesita un record ya existente Record ts, el nuevo valor val y la etiqueta del nuevo campo 1. A su vez es necesaria una prueba de que la etiqueta nueva no este repetida en las etiquetas ya existentes del record, representado por el tipo Not (ElemLabel 1 ts). Si se cumplen estas condiciones entonces es posible crear un nuevo record con el nuevo campo, el cual es reflejado en el tipo Record ((1,t):: ts), donde se agrega el par con la nueva etiqueta y el tipo del valor a la lista de tipos y etiquetas original.

#### 3.3.1. Construcción de terminos de prueba

El principal problema que ocurre al utilizar esa funcion se da por el parámetro notElem. Para poder llamar a esta funcion es necesario construir una prueba de que la nueva etiqueta a agregar al record no está repetida en el record ya existente.

Una primera opcion es generar la prueba manualmente, pero esto resulta tedioso e impráctico, ya que sería necesario construir ese termino de prueba cada vez que se llame a la funcion.

Una segunda opcion es utilizar la decidibilidad del predicado a instanciar. Al tener un tipo decidible, es posible utilizar un truco del typechecker forzando la unificacion del tipo decidible con el tipo mismo o su contradiccion. Básicamente, si uno puede generar un valor del tipo Dec p, entonces puede unificarlo con p en tiempo de compilacion, o con Not p. Si en tiempo de compilacion la unificacion falla, entonces el typechecker falla.

En este caso en particular, es necesario poder obtener un valor de tipo Dec (ElemLabel 1 ts), lo cual puede hacerse con la funcion isElemLabel.

Para forzar la unificacion, se utilizan estas funciones auxiliares:

```
getYes : (d : Dec p) ->
  case d of { No _ => (); Yes _ => p}
getYes (No _ ) = ()
getYes (Yes yes) = yes
```

```
getNo : (d : Dec p) ->
  case d of { No _ => Not p; Yes _ => ()}
getNo (No no) = no
getNo (Yes _ ) = ()
```

Se analizará el caso de getYes primero, y luego se aplicará el mismo razonamiento para getNo. getYes toma un tipo Dec p y el tipo que retorna es una computacion la cual hace pattern matching sobre el valor de Dec p. Esta computacion es ejecutada en tiempo de typechecking, y tiene dos opciones: o retorna el tipo top (), o retorna el tipo p. En tiempo de typechecking, se hace pattern matching sobre Dec p. Si la prueba es de No, entonces se retorna el valor () (que efectivamente es un valor del tipo ()). Sin embargo, si la prueba es de Yes, entonces ya se tiene un valor de tipo p, por lo cual se retorna ese. El pattern matching no solo permite tener una bifurcacion sobre cuál valor retornar, sino tambien sobre cuál es el tipo de este valor retornado, pudiendo retornar dos valores con tipos totalmente distintos.

La funcion getNo es identica, pero retornando un valor del tipo Not p. A continuacion se muestran ejemplos del uso de estas funciones:

```
okYes : Elem "L1" ["L1"]
okYes = getYes $ isElem "L1" ["L1"]

okNo : Not (Elem "L1" ["L2"])
okNo = getNo $ isElem "L1" ["L2"]

badYes : Elem "L1" ["L2"]
badYes = getYes $ isElem "L1" ["L2"]

badNo : Not (Elem "L1" ["L1"])
badNo = getNo $ isElem "L1" ["L1"]
```

En el caso de <code>okYes</code>, la funcion <code>isElem</code> es computada en tiempo de typechecking, retornando la prueba de <code>Elem</code> "L1"["L1"]. Luego <code>getYes</code> hace pattern matching sobre tal valor y encuentra que se corresponde al caso de <code>Yes</code>, por lo cual retorna ese mismo valor del tipo <code>Elem</code> "L1"["L1"]. No ocurre lo mismo para el caso de <code>badYes</code>, donde en tiempo de typechecking se retorna la prueba de <code>Not</code> (<code>Elem</code> "L1"["L2"]). Al realizar pattern matching entonces <code>getYes</code> retorna (), lo cual no puede ser unificado con <code>Elem</code> "L1"["L2"], mostrando error de typechecking.

Lo mismo ocurre de forma inversa con okno y badno.

Con este truco se puede generar una prueba automática de cualquier predicado decidible, por lo que se puede simplificar el uso de consRec. Ahora puede ser utilizado de esta forma:

```
extendedRec : Record [("Nombre", String)]
extendedRec = consRec "Nombre" "Juan"
  {notElem=(getNo $ isElemLabel "Nombre" [])} emptyRec
```

#### 3.3.2. Generación de pruebas automática

Al utilizar getYes y getNo se simplifica bastante el proceso de construccion de pruebas, pero de todas formas se necesita llamar a esas funciones manualmente. Es posible mejorar este sistema.

A continuacion se muestra un nuevo truco que utiliza el mismo concepto del anterior, donde se realiza pattern matching sobre un tipo decidible en tiempo de typechecking para unificar tipos. Sin embargo, el pattern matching se realiza en el tipo mismo y no en una funcion auxiliar.

Esto es posible gracias a este tipo y esta funcion:

```
TypeOrUnit : Dec p -> Type -> Type
TypeOrUnit (Yes yes) res = res
TypeOrUnit (No _) _ = ()

mkTypeOrUnit : (d : Dec p) -> (cnst : p -> res) ->
   TypeOrUnit d res
mkTypeOrUnit (Yes prf) cnst = cnst prf
mkTypeOrUnit (No _) _ = ()
```

El tipo TypeOrUnit permite discriminar un tipo en dos casos:

- Si Dec p incluye una prueba de p, entonces se obtiene el tipo deseado.
- Si Dec p incluye una contradiccion de p, entonces se obtiene el tipo ()

Este metodo funciona cuando se necesita unificar un tipo type y TypeOrUnit dec type. Si se tiene una prueba de p entonces la unificacion va a dar correcta, pero si no se tiene una prueba entonces va a fallar el typechecking.

 ${\tt mkTypeOrUnit}$  es el constructor de este tipo. Necesita la prueba o contradiccion de p y una funcion que construya el tipo deseado dada una prueba de p. Este constructor es utilizado solamente cuando se tiene tal prueba.

Como ejemplo, se tiene una funcion addNat que debe agregar un natural a una lista solamente si ya pertenece a esta, y de caso contrario tirar error de typechecking.

```
-- isElem : DecEq a => (x : a) -> (xs : List a) ->
-- Dec (Elem x xs)
addNat : (n : Nat) -> (ns : List Nat) ->
   TypeOrUnit (isElem n ns) (List Nat)
addNat n ns = mkTypeOrUnit (isElem n ns)
   (\isElem => Prelude.List.(::) n ns)

myListOk : List Nat
myListOk = addNat 10 [10] -- [10, 10]

myListBad1 : List Nat
myListBad1 = addNat 9 [10] -- Error de typechecking

myListBad2 : Nat -> List Nat
myListBad2 n = addNat n [10] -- Error de typechecking
```

La funcion addNat hace uso de TypeOrUnit, forzando que se cumpla el predicado Elem n ns. Si no se cumple la funcion retorna (). La unificacion se realiza en la llamada en myListOk. La llamada a addNat [10] 10 retorna el tipo TypeOrUnit (isElem 10 [10]) (List Nat), pero myListOk espera List Nat. En tiempo de typechecking se evalúa isElem 10 [10], el cual retorna una prueba de Elem 10 [10], por lo que TypeOrUnit (isElem 10 [10]) (List Nat) evalúa a List Nat, compilando correctamente el codigo.

En el caso de myListBad1, como isElem 9 [10] retorna una contradiccion de Elem 9 [10], TypeOrUnit (isElem 9 [10]) (List Nat) evalúa a (), el cual no puede ser unificado con List Nat, tirando un error de typechecking.

Otro caso de error ocurre con myListBad2. En este caso es imposible evaluar completamente isElem n [10], ya que no se conoce el valor de n en tiempo de typechecking. Por lo tanto no se puede evaluar TypeOrUnit (isElem n [10]) (List Nat), lo cual hace que falle la unificacion con List Nat.

Por lo tanto, con este metodo es posible forzar, en tiempo de compilacion, a que se cumpla un predicado específico. Si el predicado puede ser evaluado y es correcto, entonces el codigo compila correctamente. Si el predicado no puede ser evaluado, o puede pero resulta ser incorrecto, entonces el codigo no compila.

Para poder aplicar este metodo a los records, es necesario poder tener una funcion que obtenga una prueba o contradiccion de que los campos del record no tienen repetidos. Esa funcion es la siguiente:

```
ifNotSetHereThenNeitherThere : Not (IsSet xs) ->
  Not (IsSet (x :: xs))
ifNotSetHereThenNeitherThere notXsIsSet
  (IsSetCons xIsInXs xsIsSet) = notXsIsSet xsIsSet
ifIsElemThenConsIsNotSet : Elem x xs ->
  Not (IsSet (x :: xs))
ifIsElemThenConsIsNotSet xIsInXs
  (IsSetCons notXIsInXs xsIsSet) = notXIsInXs xIsInXs
isSet : DecEq t => (xs : List t) -> Dec (IsSet xs)
isSet [] = Yes IsSetNil
isSet (x :: xs) with (isSet xs)
  isSet (x :: xs) | No notXsIsSet =
    No $ ifNotSetHereThenNeitherThere notXsIsSet
  isSet (x :: xs) | Yes xsIsSet with (isElem x xs)
    isSet (x :: xs) | Yes xsIsSet | No notXInXs =
      Yes $ IsSetCons notXInXs xsIsSet
    isSet (x :: xs) | Yes xsIsSet | Yes xInXs =
      No $ ifIsElemThenConsIsNotSet xInXs
isLabelSet : DecEq lty => (ts : LabelList lty) ->
  Dec (IsLabelSet ts)
isLabelSet ts = isSet (labelsOf ts)
```

ifNotSetHereThenNeitherThere y ifIsElemThenConsIsNotSet son dos lemas necesarios para poder definir isSet. isSet toma una lista de valores que pueden chequearse por igualdad, y retorna o una prueba de que no tiene repetidos o

una prueba de que los hay.

isLabelSet simplemente permite aplicar la funcion isSet al tipo IsLabelSet.

La implementacion de isSet realiza un análisis de casos sobre el largo de la lista. Para el caso de lista vacía esta no tiene elementos repetidos por definicion. Para el caso de que tenga un elemento seguido de la cola de la lista, realiza dos análisis de casos seguidos, verificando si la cola de la lista no tiene repetidos (utilizando recursion), y luego verificando que la cabeza de la lista no pertenezca a la cola de esta. En algunos casos utiliza los lemas definidos previamente si es necesario.

En Idris un análisis de casos que tiene impacto en los tipos utiliza el identificador with. Su sintaxis es del estilo

```
func params with (expression)
  func params | Caso1 val1 = ...
  func params | Caso2 val2 = ...
```

La expression dentro del with es deconstruida en sus constructores correspondientes. La diferencia con case es que with permite redefinir los parámetros anteriores según el resultado del matcheo de la expresion. Al ser Idris un lenguaje con tipos dependientes pueden ocurrir situaciones donde una expresion matchee solamente cuando otros valores previos de la definicion tienen valores fijos. Un ejemplo es el siguiente:

```
eq : (n : Nat) -> (m : Nat) -> Bool
eq n m with (decEq n m)
  eq n n | Yes Refl = True
  eq n m | No notNEqM = False
```

El matcheo de Yes tiene un valor de tipo val: n = m. Como la única forma de que se tenga una prueba de ambos es que n sea efectivamente m, se puede unificar val con Refl: n = n y cambiar eq n m por eq n n en la definicion de la funcion.

Luego de tener la funcion de decidibilidad anterior definida, es posible aplicar el tipo TypeOrUnit a los records de la siguiente forma:

```
RecordOrUnit : DecEq lty => LabelList lty -> Type
RecordOrUnit ts = TypeOrUnit (isLabelSet ts) (Record ts)
```

RecordOrUnit ts evalúa a Record ts cuando se cumple que ts no tiene repetidos, pero evalúa a () cuando tiene repetidos.

Con este tipo es posible tener la siguiente funcion que extiende un record:

```
consRecAuto : DecEq lty => {ts : LabelList lty} ->
  {t : Type} -> (l : lty) -> (val : t) -> Record ts ->
  RecordOrUnit ((l,t) :: ts)
consRecAuto {ts} {t} l val (MkRecord tsIsLabelSet hs) =
  mkTypeOrUnit (isLabelSet ((l, t) :: ts))
  (\isLabelSet => MkRecord isLabelSet (val :: hs))
```

Esta funcion es identica a consRec, solamente que no es necesario pasar una prueba de Not (ElemLabel 1 ts). Ahora se calcula automáticamente la prueba

de IsLabelSet ((1,t) :: ts)) en tiempo de typechecking y se impacta en el tipo resultante RecordOrUnit ((1,t) :: ts).

Con esta nueva funcion es posible extender un record de la siguiente manera:

```
extendedRec1 : Record [("Nombre", String)]
extendedRec1 = consRecAuto "Nombre" "Juan" emptyRec

extendedRec2 : Record [("Edad", Nat), ("Nombre", String)]
extendedRec2 = consRecAuto "Edad" 24 extendedRec1
```

Otra funcion útil que se puede definir con este metodo es la que toma una lista heterogenea y la convierte en un record:

```
hListToRecAuto : DecEq lty => (ts : LabelList lty) ->
   HList ts -> RecordOrUnit ts
hListToRecAuto ts hs = mkTypeOrUnit (isLabelSet ts)
   (\tsIsSet => MkRecord tsIsSet hs)
```

Esta funcion permite construir records solamente indicando su lista de campos, y luego su lista de valores. Un ejemplo es el siguiente:

```
rec : Record [("Apellido", String), ("Nombre", String)]
rec = hListToRecAuto
  [("Apellido", String), ("Nombre", String)]
  ["Sanchez", "Juan"]
```

## 3.4. Proyeccion de un record

Dado un record, una funcionalidad importante que se puede tener es poder filtrar el record dada una lista de etiquetas, donde es posible obtener un nuevo record con solo los campos indicados en una lista de etiquetas proporcionada por el usuario. Esto se llama proyeccion de un record, y es otra funcionalidad clave de los records extensibles.

Un ejemplo de su uso seria el siguiente

hProjectByLabelsAuto en este caso toma una lista de etiquetas y retorna un subconjunto del record con solo esas etiquetas.

El uso de \$ permite aplicar una funcion a sus argumentos sin tener que usar parentesis. Esta sintaxis proviene de Haskell y es definida de esta manera:

```
(\$): (a -> b) -> a -> b
(\$) f a = f a
```

Esta herramienta facilita la aplicacion de funciones gracias a su precedencia, donde  $f \times (g \times (h \times z))$  es equivalente a  $f \times g \times g \times h \times z$ .

Para comenzar a definir proyeccion sobre records se necesita definir un predicado que indique que una lista es el resultado de eliminar un elemento de otra.

```
data DeleteElemPred : (xs : List t) -> Elem x xs ->
  List t -> Type where
  DeleteElemPredHere : DeleteElemPred (x :: xs) Here xs
  DeleteElemPredThere : {isThere : Elem y xs} ->
    DeleteElemPred xs isThere ys ->
    DeleteElemPred (x :: xs) (There isThere) (x :: ys)
```

El tipo DeleteElemPred xs elem ys toma dos listas xs y ys, una prueba elem de que un elemento pertenece a xs, y representa el predicado de que la lista ys es el resultado de eliminar ese elemento de xs. El caso base del predicado es DeleteElemPredHere, donde el elemento a eliminar es el primero de la lista. El caso recursivo es DeleteElemPredThere, donde elimina el elemento del resto de la lista y deja el primer elemento sin cambiar.

La función que computa la lista resultado que es equivalente al predicado anterior es

```
deleteElem : (xs : List t) -> Elem x xs -> List t
deleteElem (x :: xs) Here = xs
deleteElem (x :: xs) (There inThere) =
  let rest = deleteElem xs inThere
  in x :: rest
```

En un lenguaje como Idris se pueden definir funciones que computen un resultado, o un predicado que establezcan que uno de sus parametros sea ese mismo resultado.

Estas dos formas de definir una funcionalidad son equivalentes, lo cual se prueba definiendo las siguientes dos funciones:

```
fromDeleteElemPredToComp : {xs1, xs2 : List t} ->
    {isElem : Elem x xs1} -> DeleteElemPred xs1 isElem xs2 ->
    xs2 = deleteElem xs1 isElem

fromCompToDeleteElemPred : (xs : List t) ->
    (isElem : Elem x xs) ->
    DeleteElemPred xs isElem (deleteElem xs isElem)
```

Estas funciones permiten establecer un isomorfismo entre deleteElem xs isElem y DeleteElemPred xs isElem ys.

Para completar la definicion del tipo se define la funcion de decibilidad de esta, con el siguiente tipo

```
isDeleteElemPred : DecEq t => (xs : List t) ->
  (isElem : Elem x xs) -> (res : List t) ->
  Dec (DeleteElemPred xs isElem res)
```

Con esta definicion se pueden definir los predicados de proyeccion

```
data IsProjectLeft : DecEq lty => List lty -> LabelList lty ->
  LabelList lty -> Type where
  IPL_EmptyLabels : DecEq lty => IsProjectLeft {lty} [] ts []
  IPL_EmptyVect : DecEq lty => IsProjectLeft {lty} ls []
  IPL_ProjLabelElem : DecEq lty => (isElem : Elem l ls) ->
    DeleteElemPred ls isElem lsNew ->
    IsProjectLeft {lty} lsNew ts res1 ->
    IsProjectLeft ls ((1,ty) :: ts) ((1,ty) :: res1)
  IPL_ProjLabelNotElem : DecEq lty => Not (Elem l ls) ->
    IsProjectLeft {lty} ls ts res1 ->
    IsProjectLeft ls ((1,ty) :: ts) res1
data IsProjectRight : DecEq lty => List lty -> LabelList lty ->
  LabelList lty -> Type where
  IPR_EmptyLabels : DecEq lty => IsProjectRight {lty} [] ts ts
  IPR_EmptyVect : DecEq lty => IsProjectRight {lty} ls [] []
  IPR_ProjLabelElem : DecEq lty => (isElem : Elem l ls) ->
   DeleteElemPred ls isElem lsNew ->
   IsProjectRight {lty} lsNew ts res1 ->
   IsProjectRight ls ((1,ty) :: ts) res1
  IPR_ProjLabelNotElem : DecEq lty => Not (Elem l ls) ->
    IsProjectRight {lty} ls ts res1 ->
    IsProjectRight ls ((1,ty) :: ts) ((1,ty) :: res1)
```

IsProjectLeft 1s ts1 ts2 indica que la proyeccion de las etiquetas 1s sobre la lista ts1 equivale a la lista ts2. IsProjectRight 1s ts1 ts2 indica que al proyectar las etiquetas 1s sobre la lista ts1, las etiquetas que no pertenecen a la proyeccion son las de la lista ts2. Se le llama proyeccion por izquierda y derecha ya que dada una lista de etiquetas esta siempre se puede dividir en dos al ser proyectada por otra, donde en la izquierda estan las etiquetas que pertenecen a la lista a proyectar y en la derecha las que no.

Algunos ejemplos son los siguientes

```
IsProjectLeft ["Edad"] [("Edad", Nat),("Nombre",String)]
  [("Edad",Nat)]
IsProjectRight ["Edad"] [("Edad", Nat),("Nombre",String)]
  [("Nombre", String)]
```

Al momento de definir estos tipos, el caso de IsProjectLeft tiene estos posibles casos

■ Si se intenta proyectar por una lista vacia, la proyeccion de esta sobre cualquier otra lista va a ser vacia tambien, indicado por el tipo IsProjectLeft lty

- Si se intenta proyectar hacia una lista vacia, entonces el resultado va a ser vacio tambien, indicado por el tipo IsProjectLeft lty ls [] []
- Si la lista donde se proyecta es del estilo (1,ty) :: ts, y si se tiene que esa etiqueta pertenece a la otra lista, con isElem : Elem 1 ls, entonces para proyectar la otra lista se debe eliminar ese elemento de esa con DeleteElemPred ls isElem lsNew obteniendo una nueva lista lsNew sin ese elemento. Luego se debe proyectar el resto de la lista con IsProjectLeft lty lsNew ts res1. La proyeccion final equivale a ese resultado mas el elemento eliminado, con el tipo IsProjectLeft ls ((1,ty) :: ts) ((1,ty) :: res1)
- Si la lista donde se proyecta es del estilo (1,ty) :: ts, y si se tiene que esa etiqueta no pertenece a la otra lista, con Not (Elem 1 ls), entonces la proyeccion equivale a proyectar sobre el resto de la lista de la forma IsProjectLeft lty ls ts res1, retornando IsProjectLeft ls ((1,ty) :: ts) res1

Para el caso de IsProjectRight los casos son similares, donde la diferencia cae en que si se va a proyectar por un elemento que existe en la lista origen, entonces no se agrega a la lista destino (y si no pertenece a la lista origen si se agrega a la lista destino).

Al igual que DeleteElemPred, se puede definir una funcion que compute la proyeccion misma. Para el caso de IsProjectLeft (no vamos a enforcarnos en IsProjectRight de ahora en mas) esta es

```
projectLeft : DecEq lty => List lty -> LabelList lty ->
   LabelList lty
projectLeft [] ts = []
projectLeft ls [] = []
projectLeft ls ((1,ty) :: ts) with (isElem 1 ls)
   projectLeft ls ((1,ty) :: ts) | Yes lIsInLs =
   let delLFromLs = deleteElem ls lIsInLs
        rest = projectLeft delLFromLs ts
   in (1,ty) :: rest
   projectLeft ls ((1,ty) :: ts) | No _ = projectLeft ls ts
```

La implementacion de esta funcion sigue los mismos pasos de la definicion del predicado  ${\tt IsProjectLeft}.$ 

Tambien se puede definir un isomorfismo entre projectLeft ls ts1 y IsProjectLeft ls ts1 ts2 de esta forma

```
fromIsProjectLeftToComp : DecEq lty =>
  {ls : List lty} -> {ts1, ts2 : LabelList lty} ->
  IsProjectLeft ls ts1 ts2 ->
  IsSet ls -> ts2 = projectLeft ls ts1

fromCompToIsProjectLeft : DecEq lty => (ls : List lty) ->
  (ts : LabelList lty) ->
  IsProjectLeft ls ts (projectLeft ls ts)
```

En el caso de fromIsProjectLeftToComp es necesario tener una prueba de IsSet ls. Como las etiquetas de un record no pueden repetirse, no tiene sentido intentar proyectar por una lista con etiquetas repetidas. Sin embargo, en la definicion de varios tipos y funciones no es necesario explicitar esa precondicion de que no tengan etiquetas repetidas (proporcionando una prueba de IsSet ls). Solo es necesario proporcionar esa prueba en fromIsProjectLeftToComp porque necesita esa prueba en su implementacion. Su implementacion completa se encuentra en el apendice.

Dados estos tipos y funciones es posible definir una funcion que particione una lista heterogenea en dos, donde en el lado izquierdo se tienen los valores asociados a las etiquetas proyectadas y en el lado derecho los valores no asociados a tales

```
hProjectByLabelsHList : DecEq lty =>
  {ts : LabelList lty} -> (ls : List lty) ->
  HList ts ->
  ((ts1 : LabelList lty **
    (HList ls1, IsProjectLeft ls ts ts1)),
  (ts2 : LabelList lty **
    (HList ls2, IsProjectRight ls ts ts2)))
hProjectByLabelsHList [] {ts} hs =
  (([] ** ([], IPL_EmptyLabels)),
  (ts ** (hs, IPR_EmptyLabels)))
hProjectByLabelsHList _ [] =
  (([] ** ([], IPL_EmptyVect)),
  ([] ** ([], IPR_EmptyVect)))
hProjectByLabelsHList {lty} ls ((::) {l=12} {t}
    \{ts=ts2\}\ val\ hs) =
  case (isElem 12 ls) of
    Yes l2InLs =>
      let
        (lsNew ** isDelElem) = deleteElemPred ls 12InLs
        ((subInLs ** (subInHs, subPrjLeft)),
          (subOutLs ** (subOutHs, subPrjRight))) =
          hProjectByLabelsHList {lty=lty} {ts=ts2} lsNew hs
        rPrjRight = IPR ProjLabelElem {l=12} {ty=t} {ts=ts2}
          {res1=subOutLs} 12InLs isDelElem subPrjRight
        rPrjLeft = IPL_ProjLabelElem {l=12} {ty=t} {ts=ts2}
          {res1=subInLs} 12InLs isDelElem subPrjLeft
        rRight = (subOutLs ** (subOutHs, rPrjRight))
        rLeft = ((12,t) :: subInLs ** (
          (::) {l=12} val subInHs, rPrjLeft))
         (rLeft, rRight)
    No notL2InLs =>
      let
        ((subInLs ** (subInHs, subPrjLeft)),
          (subOutLs ** (subOutHs, subPrjRight))) =
          hProjectByLabelsHList {lty=lty} {ts=ts2} ls hs
        rPrjLeft = IPL_ProjLabelNotElem {l=12} {ty=t} {ts=ts2}
```

```
{res1=subInLs} notL2InLs subPrjLeft
rPrjRight = IPR_ProjLabelNotElem {l=l2} {ty=t} {ts=ts2}
    {res1=subOutLs} notL2InLs subPrjRight
rLeft = (subInLs ** (subInHs, rPrjLeft))
rRight = ((12,t) :: subOutLs ** (
    (::) {l=l2} val subOutHs, rPrjRight))
in
    (rLeft, rRight)
```

Se analizara esta funcion en partes:

```
hProjectByLabelsHList : DecEq lty => {ts : LabelList lty} ->
  (ls : List lty) -> HList ts ->
  ((ts1 : LabelList lty **
     (HList ts1, IsProjectLeft ls ts ts1)),
  (ts2 : LabelList lty **
     (HList ts2, IsProjectRight ls ts ts2)))
```

La definicion de la funcion toma una lista de etiquetas a proyectar ls: List lty, la lista heterogenea con valores <code>HList ts</code>, y retorna dos valores. En la izquierda retorna una lista tsl: LabelList lty junto a la lista heterogenea <code>HList tsl</code>. A su vez retorna una prueba de <code>IsProjectLeft ls ts tsl</code>, que indica que al proyectar ls sobre ts el resultado fue tsl, es decir, tsl contiene las etiquetas proyectadas y <code>HList tsl</code> contiene los valores de esas etiquetas. El caso de la derecha es similar, pero teniendo una prueba de <code>IsProjectRight ls tstsl</code>, lo cual indica que <code>HList tsl</code> tiene los valores de las etiquetas que no estan en ls.

```
hProjectByLabelsHList [] {ts} hs =
  (([] ** ([], IPL_EmptyLabels)),
  (ts ** (hs, IPR_EmptyLabels)))
```

Si se intenta proyectar por una lista vacia, entonces en la izquierda se retorna vacio pero en la derecha la HList original. Para las pruebas se utilizan las que corresponden a este caso base.

```
hProjectByLabelsHList _ [] =
  (([] ** ([], IPL_EmptyVect)),
  ([] ** ([], IPR_EmptyVect)))
```

Si la lista heterogenea origen es vacia, entonces ambos resultados van a ser vacios. Las pruebas utilizadas son las correspondientes a este caso base tambien.

Para el caso de que la HList sea del estilo val :: hs (se utiliza la notacion (::) val hs para poder utilizar las variables implicitas 12 y t) se verifica si la

etiqueta de ese primer valor pertenece a las que se quieren proyectar, con isElem 12 ls : Elem 12 ls.

```
let
  (lsNew ** isDelElem) = deleteElemPred ls l2InLs
  ((subInLs ** (subInHs, subPrjLeft)),
        (subOutLs ** (subOutHs, subPrjRight))) =
        hProjectByLabelsHList {lty=lty} {ts=ts2} lsNew hs
    rPrjRight = IPR_ProjLabelElem {l=l2} {ty=t} {ts=ts2}
        {res1=subOutLs} l2InLs isDelElem subPrjRight
    rPrjLeft = IPL_ProjLabelElem {l=l2} {ty=t} {ts=ts2}
        {res1=subInLs} l2InLs isDelElem subPrjLeft
    rRight = (subOutLs ** (subOutHs, rPrjRight))
    rLeft = ((l2,t) :: subInLs ** (
        (::) {l=l2} val subInHs, rPrjLeft))
    in
        (rLeft, rRight)
```

Si pertenece a la lista, entonces se utiliza la siguiente funcion para sacar esa etiqueta de las que se quieren proyectar

```
deleteElemPred : {x : t} -> (xs : List t) ->
  (elem : Elem x xs) ->
  (res : List t ** DeleteElemPred xs elem res)
```

Al llamar a deleteElemPred 1s 12InLs se obtiene el predicado correspondiente a tal eliminacion. Luego se llama recursivamente a hProjectByLabelsH-List con esa nueva lista en hProjectByLabelsHList {lty=lty} {ts=ts2} lsNew hs. Con IPR\_ProjLabelElem {l=12} {ty=t} {ts=ts2} {res1=subOutLs} 12InLs isDelElem subPrjRight se genera la prueba de IsProjectRight correspondiente, y con IPL\_ProjLabelElem {l=12} {ty=t} {ts=ts2} {res1=subInLs} 12InLs isDelElem subPrjLeft la prueba de IsProjectLeft. Luego se retorna en la derecha (subOutLs \*\* (subOutHs, rPrjRight)) que no contiene el valor actual y ((12,t) :: subInLs \*\* ((::) {l=12} val subInHs, rPrjLeft)) que si contiene el valor actual (de nuevo se utiliza el constructor de forma prefija para asignar las variables implicitas correspondientes).

```
No notL2InLs =>
let
   ((subInLs ** (subInHs, subPrjLeft)),
        (subOutLs ** (subOutHs, subPrjRight))) =
        hProjectByLabelsHList {lty=lty} {ts=ts2} ls hs
   rPrjLeft = IPL_ProjLabelNotElem {l=l2} {ty=t} {ts=ts2}
        {res1=subInLs} notL2InLs subPrjLeft
   rPrjRight = IPR_ProjLabelNotElem {l=l2} {ty=t} {ts=ts2}
        {res1=subOutLs} notL2InLs subPrjRight
   rLeft = (subInLs ** (subInHs, rPrjLeft))
   rRight = ((l2,t) :: subOutLs ** (
        (::) {l=l2} val subOutHs, rPrjRight))
```

Si no pertenece a la lista, entonces simplemente se realiza la proyeccion sobre el resto de los campos y se agrega ese valor a la derecha. La proyeccion se realiza con hProjectByLabelsHList {lty=lty} {ts=ts2} ls hs, la prueba de IsProjectLeft se genera con IPL\_ProjLabelNotElem {l=l2} {ty=t} {ts=ts2} {res1=subInLs} notL2InLs subPrjLeft, y la prueba de IsProjectRight se genera con IPR\_ProjLabelNotElem {l=l2} {ty=t} {ts=ts2} {res1=subOutLs} notL2InLs subPrjRight. El valor izquierdo devuelto es (subInLs \*\* (subInHs, rPrjLeft)) y el derecho ((12,t) :: subOutLs \*\* ((::) {l=l2} val subOutHs, rPrjRight)), donde se agrega el valor actual.

Esta funcion permite proyectar sobre listas heterogeneas, pero se necesita proyectar sobre records extensibles. Para ello es necesario tener una garantia de que las etiquetas resultantes de la proyeccion no sean repetidas. Eso se logra con el siguiente lema:

```
hProjectByLabelsLeftIsSet_Lemma2 : DecEq lty =>
  {ls : List lty} -> {ts1, ts2 : LabelList lty} ->
  IsProjectLeft ls ts1 ts2 -> IsLabelSet ts1 ->
  IsLabelSet ts2
```

Este lema indica que si las etiquetas de ts1 no son repetidas, representadas con una prueba de IsLabelSet ts1, y se realiza la proyeccion con IsProjectLeft ls ts1 ts2, entonces las etiquetas resultantes no son repetidas, representadas con una prueba de IsLabelSet ts2.

Con este lema se tienen todas las herramientas para definir la proyeccion sobre un Record

```
hProjectByLabelsWithPred : DecEq lty =>
  {ts1 : LabelList lty} -> (ls : List lty) ->
  Record ts1 -> IsSet ls ->
  (ts2 : LabelList lty **
      (Record ts2, IsProjectLeft ls ts1 ts2))
hProjectByLabelsWithPred ls rec lsIsSet =
  let
    isLabelSet = recLblIsSet rec
    hs = recToHList rec
    (lsRes ** (hsRes, prjLeftRes)) =
      fst $ hProjectByLabelsHList ls hs
    isLabelSetRes = hProjectByLabelsLeftIsSet_Lemma2
      prjLeftRes isLabelSet
  in (lsRes **
    (hListToRec {prf=isLabelSetRes} hsRes, prjLeftRes))
```

Eltipo hProjectByLabelsWithPred : DecEq lty =>{ts1 : LabelList lty} -> (ls : List lty) ->Record ts1 ->IsSet ls -> (ts2 : LabelList lty \*\* (Record ts2, IsProjectLeft ls ts1 ts2)) toma un record Record ts1, una lista de etiquetas ls no repetidas (con una prueba de IsSet ls), y retorna un nuevo record Record ts2 con una prueba de que ts2 es una proyeccion, con IsProjectLeft ls ts1 ts2.

La implementacion toma la lista heterogenea del record y llama a hProjectByLabelsHList ls hs obteniendo el nuevo HList y la prueba IsProjectLeft ls ts1 ts2. Sabiendo que rec es un record se sabe que IsLabelSet ts1, por lo que, utilizando el lema hProjectByLabelsLeftIsSet\_Lemma2, se sabe que IsLabelSet ts2. Con esos dos datos es posible crear el nuevo record Record ts2.

Para no tener que pasar una prueba manual de IsSet ls se define la siguiente funcion, utilizando el mismo truco definido anteriormente.

```
hProjectByLabelsWithPredAuto : DecEq lty =>
  {ts1 : LabelList lty} -> (ls : List lty) ->
  Record ts1 -> TypeOrUnit (isSet ls)
  ((ts2 : LabelList lty **
        (Record ts2, IsProjectLeft ls ts1 ts2)))
hProjectByLabelsWithPredAuto ls rec =
   mkTypeOrUnit (isSet ls)
  (\isSet => hProjectByLabelsWithPred ls rec isSet)
```

Esta funcion utiliza el tipo TypeOrUnit para encapsular un predicado decidible que se debe calcular en tiempo de compilacion. A diferencia de RecordOrUnit, cuyo predicado era IsLabelSet ts, en este caso el predicado a calcular es IsSet ls.

TypeOrUnit (isSet ls) rest esconde la necesidad de calcular isSet ls : IsSet ls a mano y le permite a rest utilizar esa prueba.

#### 3.4.1. Computacion de funcion en el tipo del record

Esta definicion es suficiente para realizar la proyeccion izquierda de una lista de etiquetas, pero resulta dificil de utilizar en la practica. Esta funcion retorna una lista de etiquetas arbitraria, el record resultante y una prueba de que es la proyeccion, pero no retorna ninguna otra informacion que permita a uno utilizar ese record directamente.

Su uso seria el siguiente

Al tener el valor projRec no es posible conocer el valor de ts2 en tiempo de compilacion, cuando pareceria obvio que es [('Apellido', String),

('Edad', Nat)]. Para tener informacion sobre esa lista es necesario hacer un analisis de casos sobre la prueba de IsProjectLeft.

Esto resulta util cuando se utiliza en otras funciones que necesitan realizar ese analisis de casos, o tienen en su alcance o contexto otras pruebas y tipos que pueden hacer uso de ese predicado. Pero para la mayoria de sus usos quisieramos utilizarlo de la forma que se indico al comienzo de esta seccion

```
projectedRec : Record [("Apellido", String),
    ("Edad", Nat)]
projectedRec = hProjectByLabelsAuto
    ["Apellido", "Edad"] rec
```

Para ello es necesario computar la lista resultante en el tipo mismo del record. Es decir, se necesita que en tiempo de compilacion la funcion de proyeccion tome la lista de etiquetas del record, la lista de etiquetas que se quieren proyectar, y en el tipo del record compute tal proyeccion y retorne la lista resultante.

Esto es posible gracias a estas dos funciones definidas anteriormente

```
projectLeft : DecEq lty => List lty -> LabelList lty ->
   LabelList lty

fromIsProjectLeftToComp : DecEq lty =>
   {ls : List lty} -> {ts1, ts2 : LabelList lty} ->
   IsProjectLeft ls ts1 ts2 ->
   IsSet ls -> ts2 = projectLeft ls ts1
```

projectLeft permite computar la lista resultante en el tipo, y fromIsProjectLeftToComp permite tomar una prueba del predicado IsProjectLeft ls ts1 ts2 y asegurar que su resultado sea identico a la computacion projectLeft ls ts1. Con estas funciones se puede definir la siguiente funcion

```
hProjectByLabels : DecEq lty =>
  {ts : LabelList lty} -> (ls : List lty) ->
  Record ts -> IsSet ls ->
  Record (projectLeft ls ts)
hProjectByLabels {ts} ls rec lsIsSet =
  let
    isLabelSet = recLblIsSet rec
   hs = recToHList rec
    (lsRes ** (hsRes, prjLeftRes)) =
      fst $ hProjectByLabelsHList ls hs
    isLabelSetRes = hProjectByLabelsLeftIsSet_Lemma2
      prjLeftRes isLabelSet
    resIsProjComp =
      fromIsProjectLeftToComp prjLeftRes lsIsSet
    recRes = hListToRec {prf=isLabelSetRes} hsRes
  in rewrite (sym resIsProjComp) in recRes
```

La funcion es similar a hProjectByLabelsWithPred, pero en vez de retornar (ts2 : LabelList lty \*\* (Record ts2, IsProjectLeft ls

ts1 ts2)) realiza la computacion de la lista resultado en el tipo mismo y la retorna en Record (projectLeft ls ts).

Su implementacion es identica a hProjectByLabelsWithPred, realizando la proyeccion sobre la lista heterogenea con hProjectByLabelsHList y obteniendo una prueba de etiquetas no repetidas con hProjectByLabelsLeftIsSet\_Lemma2. Sin embargo, como la proyeccion sobre HList retorno una prueba de IsProjectLeft ls ts ts2 en prjLeftRes, se puede llamar a fromIsProjectLeftToComp prjLeftRes lsIsSet para tener la igualdad ts2 = projectLeft (ls ts). La llamada a hListToRec {prf=isLabelSetRes} hsRes retorna un record utilizando esa misma lista ts2 (ya que isLabelSetRes es una prueba de IsLabelSet ts2). Por lo tanto, teniendo esa igualdad se puede utilizar la herramienta rewrite (sym resIsProjComp) in recRes para modificar el tipo de recRes de Record ts2 a Record (projectLeft ls ts).

In Idris, rewrite proof in rest toma una prueba de igualdad proof (projectLeft (ls ts) = ts2 en este caso) y realiza una sustitucion del termino de la derecha con el de la izquierda en rest (recRes en este caso).

Siguiendo con la estrategia de hProjectByLabelsWithPredAuto, se define otra funcion que evita tener que calcular IsSet ls a mano

```
hProjectByLabelsAuto : DecEq lty =>
  {ts : LabelList lty} -> (ls : List lty) ->
  Record ts -> TypeOrUnit (isSet ls)
  (Record (projectLeft ls ts))
hProjectByLabelsAuto {ts} ls rec =
  mkTypeOrUnit (isSet ls)
  (\lsIsSet => hProjectByLabels {ts=ts} ls rec lsIsSet)
```

Esta implementacion es identica a hProjectByLabelsWithPredAuto.

Con estas nuevas funciones, por fin es posible realizar la proyeccion de la siguiente manera

```
projectedRec: Record [("Apellido", String),
    ("Edad", Nat)]
projectedRec = hProjectByLabelsAuto
    ["Apellido", "Edad"] rec
```

## 3.5. Busqueda de un elemento en un record

En la seccion anterior se definio la proyeccion de un record dada una lista de etiquetas. Sin embargo falta definir otra funcionalidad muy importante: Obtener el valor de un record dada una etiqueta.

La funcionalidad de lookup se puede definir de forma muy similar a las anteriores. Primero se comienza con un predicado que indica que una lista de etiquetas contiene a otra etiqueta con un determinado tipo en particular

```
data HasField : (l : lty) -> LabelList lty ->
  Type -> Type where
  HasFieldHere : HasField l ((l,ty) :: ts) ty
```

```
HasFieldThere : HasField l1 ts ty1 ->
HasField l1 ((l2,ty2) :: ts) ty1
```

HasField 1 ts ty indica que en la lista de etiquetas ts existe la etiqueta 1 que tiene asociado el tipo ty. Un ejemplo de tal tipo es HasField 'Edad' [('Nombre', String), ('Edad', Nat)] Nat. Su definicion es recursiva, donde HasFieldHere indica que la etiqueta se encuentra en la cabeza de la lista, y HasFieldThere indica que se encuentra en la cola de la lista.

Una primera definicion es la de obtener el valor de una lista heterogenea:

```
hLookupByLabel_HList : DecEq lty => {ts : LabelList lty} ->
  (l : lty) -> HList ts -> HasField l ts ty -> ty
hLookupByLabel_HList _ (val :: _) HasFieldHere = val
hLookupByLabel_HList l (_ :: ts)
  (HasFieldThere hasFieldThere) =
hLookupByLabel_HList l ts hasFieldThere
```

Esta funcion toma una etiqueta l: lty, una lista heterogenea HList ts, y una prueba de que esa etiqueta pertenece a esa lista con HasField lts ty. Con tales dato permite retornar un valor del tipo ty, el cual es el asociado a tal etiqueta.

Su implementacion realiza pattern matching sobre el predicado, obteniendo el valor val: ty si la etiqueta esta en la cabeza de la lista o realizando un llamado recursivo sino.

La funcion que obtiene un elemento de un record es sencilla

```
hLookupByLabel : DecEq lty => {ts : LabelList lty} ->
  (l : lty) -> Record ts -> HasField l ts ty -> ty
hLookupByLabel {ts} {ty} l rec hasField =
  hLookupByLabel_HList {ts} {ty} l (recToHList rec) hasField
```

Como la definicion de record de este trabajo contiene la lista heterogenea dentro de el, entonces esta funcion simplemente aplica la funcion definida anteriormente a tal lista heterogenea del record.

Al igual que las funcionalidades anteriores, se puede definir una funcion que calcule el predicado de forma automatica en tiempo de compilacion. Sin embargo, con el caso de lookup ocurre un problema. Idealmente, uno quisiera tener la siguiente funcion:

```
hasField : Deceq lty => (l : lty) ->
  (ts : LabelList lty) -> (ty : Type) ->
  Dec (HasField l ts ty)
```

Con esta funcion uno podria definir la funcion de calculo automatico de la siguiente forma

```
hLookupByLabelAuto : DecEq lty => {ts : LabelList lty} ->
  (1 : lty) -> Record ts ->
  TypeOrUnit (hasField l ts ty) ty
hLookupByLabelAuto {ts} {ty} l rec =
```

```
mkTypeOrUnit (hasField l ts ty)
  (\tsHasL => hLookupByLabel {ts} {ty} l rec tsHasL)
```

El problema ocurre que no es posible definir hasField.

Para implementar hasField, se tiene una variable ty: Type que puede ser cualquier tipo. A su vez, si la etiqueta 1: 1ty pertenece a la lista, entonces en la lista la etiqueta va a tener asociado un tipo ty2: Type, que puede ser cualquier otro. El problema ocurre en que no se puede verificar la igualdad de los tipos ty y ty2, ya que no existe forma de igualar valores de tipo Type.

Por ejemplo, se puede tener hasField 'Edad' [('Edad', Nat)] Nat o hasField 'Edad' [('Edad', Nat)] String. El primer caso deberia retornar Yes con una prueba, mientras que el segundo deberia retornar No con una prueba de su opuesto, ya que el tipo Nat y String son distintos, por lo que el predicado falla. Sin embargo, no es posible realizar un chequeo Nat = Nat o Not (Nat = String), ya que eso requeriria que existiese una instancia de DecEq Type, la cual no existe y no es posible definir.

El hecho de que no existe una instancia DecEq Type en el lenguaje tiene varias razones:

- Dados dos tipos cualquiera en un lenguaje de tipos dependientes, verificar si son iguales no es decidible. Es decir, teoricamente, es imposible crear una instancia de DecEq Type
- Si existiera, no existiria el polimorfismo parametrico. El polimorfismo parametrico permite abstraerse de un tipo y trabajar con el sin saber especificamente cual es. Esto permite a uno deducir propiedades de la funcion solamente conociendo el tipo. Por ejemplo, si se tiene la funcion id: a ->a, se puede deducir que la unica posible implementacion es id val = val, ya que es imposible que la funcion pueda conocer informacion sobre el tipo a : Type.

Sin embargo, si existiera una instancia de DecEq Type, entonces la siguiente funcion seria valida

```
id2 : a -> a
id2 {a=a} val with (decEq a Nat)
  id2 {a=Nat} val | Yes aIsNat = 10
  id2 {a=a} val | No notAIsNat = val
```

La funcion id2: a ->a retorna el mismo valor para todos los tipos, menos para el tipo Nat, donde retorna siempre el valor 10. Esta funcion contradice las garantias de parametricidad del tipo a ->a.

A pesar de no poder utilizar TypeOrUnit, es posible definir una funcion que calcule el predicado de forma automatica utilizando la funcionalidad auto de Idris, de esta forma

```
hLookupByLabelAuto : DecEq lty => {ts : LabelList lty} ->
  (1 : lty) -> Record ts ->
  {auto hasField : HasField l ts ty} -> ty
hLookupByLabelAuto {ts} {ty} l rec {hasField} =
  hLookupByLabel_HList {ts} {ty} l (recToHList rec) hasField
```

auto hasField: HasField l ts ty indica que el typechecker de ldris va a intentar generar el predicado HasField l ts ty de forma automatica. Idris se da cuenta de que la definicion de HasField es inductiva y tiene distintos constructores, por lo que intenta, mediante fuerza bruta, aplicar los constructores secuencialmente hasta encontrar un valor que tenga el tipo deseado.

Por ejemplo, si se quisiera obtener una prueba de HasField 'Edad' [('Nombre', String), ('Edad', Nat)] Nat, Idris primero va a construir HasFieldHere. Como HasFieldHere no puede unificarse con el tipo deseado, sigue con el siguiente caso. Proximamente, Idris intenta construir HasFieldThere HasFieldHere. Idris consigue unificar el tipo de ese valor con el esperado, entonces proporciona el valor HasFieldThere HasFieldHere a la funcion hLookupByLabelAuto.

Esta funcion hLookupByLabel se utiliza de la siguiente forma:

Como conclusion, utilizar TypeOrUnit es muy util para definir estas funciones con calculo automatico, pero solo puede usarse si el predicado no depende de un valor a : Type. Si es asi, se deben utilizar funcionalidades propias del lenguaje como auto.

## Caso de estudio

#### 4.1. Introducción

Se decidió aplicar la previa definición de records extensibles en un caso de estudio. Éste consiste en definir un pequeño lenguaje aritmético con variables y constantes, y utilizar records extensibles para poder interpretarlo.

Este lenguaje *Exp* se define de la siguiente manera:

- Si *n* es un natural, entonces *n* pertenece a *Exp* como un literal.
- Si x es un string, entonces x pertenece a Exp como una variable.
- Si e1 y e2 pertenecen a Exp, entonces e1 + e2 pertenece a Exp como una suma
- Si x es un string, n es un natural y e pertenece a Exp, entonces let x := n in
  e pertenece a Exp como una sustitución de una variable.

Este lenguaje permite definir expresiones aritméticas donde se pueden definir valorse numéricos, variables, o sumar expresiones. Los siguientes son ejemplos de expresiones de este lenguaje:

```
x
x + 3
let x := 3 in x + 3
let y := 10 in (let x := 3 in x + 3)
```

En este trabajo se va a implementar tal lenguaje en Idris utilizando tipos dependientes, y se van a utilizar records extensibles para definir el contexto necesario para interpretar una expresión de este lenguaje.

### 4.2. Definición de una expresión

Una expresión de este lenguaje se define en Idris de la siguiente forma:

```
data VarDec : String -> Type where
  (:=) : (var : String) -> Nat -> VarDec var

data Exp : List String -> Type where
Add : Exp fvs1 -> Exp fvs2 ->
    IsLeftUnion_List fvs1 fvs2 fvsRes ->
    Exp fvsRes
Var : (1 : String) -> Exp [1]
Lit : Nat -> Exp []
Let : VarDec var -> Exp fvsInner ->
    DeleteLabelAtPred_List var fvsInner fvsOuter ->
    Exp fvsOuter
```

Un valor de tipo  $VarDec \ x$  contiene la variable x y un natural. Representa la declaración de una variable, como 'x' := 10 :  $VarDec \ 'x'$ .

El tipo Exp se define como un tipo parametrizado por una lista de variables. Esta lista representa la lista de variables libres que ocurren en la expresión. Un valor de tipo Exp ['x', 'y'] es una expresión que tiene a 'x' e 'y' como variables libres. Se decidió incluir la información de variables libres en el tipo para poder facilitar tanto la implementación del intérprete, como la construcción de expresiones.

Cada constructor de Exp representa cada ítem de la definición del lenguaje:

- Lit define una literal numérica. La expresión resultante es Exp [] porque no contiene variables libres.
- Var define una variable. La expresión resultante contiene esa variable como libre.
- Add define la suma de dos expresiones. IsLeftUnion\_List fvs1 fvs2 fvsRes es un predicado que indica que fvsRes es la unión por izquierda de la lista fvs1 y fvs2. Por lo tanto la expresión resultante tiene como variables libres la unión por izquierda de las variables libres de la primera expresión y de la segunda.
- Let define la sustitución de una variable. Toma una declaración de variable del estilo 'x' := 10 y una subexpresión. DeleteLabelAtPred\_List var fvsInner fvsOuter es un predicado que indica que la lista fvsOuter es el resultado de tomar la lista fvsInner y eliminarle la variable var (si var no existe en fvsInner entonces la lista resultante es idéntica a ésta). Esto significa que las variables libres de una sustitución son las variables libres de la subexpresión menos la variable que ahora está ligada.

#### 4.2.1. Construcción de una expresión

Para poder construir expresiones de este lenguaje, se definieron funciones auxiliares que construyen valores del tipo Exp.

```
var : (1 : String) -> Exp [1]
var l = Var l
lit : Nat -> Exp []
```

```
lit n = Lit n

add : Exp fvs1 -> Exp fvs2 -> Exp (leftUnion fvs1 fvs2)
add {fvs1} {fvs2} e1 e2 = Add e1 e2
   (fromLeftUnionFuncToPred {ls1=fvs1} {ls2=fvs2})

eLet : VarDec var -> Exp fvs -> Exp (deleteAtList var fvs)
eLet {var} {fvs} varDec e = Let varDec e
   (fromDeleteLabelAtListFuncToPred {l=var} {ls=fvs})
```

Las cuatro funciones simplemente realizan la aplicación del constructor correspondiente. add y elet tienen la particularidad de que realizan las computaciones sobre las listas de variables, y construyen los predicados correspondientes dada esas computaciones. leftUnion y deleteAtList son las computaciones análogas a IsleftUnion\_List y DeleteAtLabel\_List, cumpliendo con los siguientes lemas:

```
fromDeleteLabelAtListFuncToPred : DecEq lty => {l : lty} ->
   {ls : List lty} ->
   DeleteLabelAtPred_List l ls (deleteAtList l ls)
fromLeftUnionFuncToPred : DecEq lty => {ls1, ls2 : List lty} ->
   IsLeftUnion_List {lty} ls1 ls2 (leftUnion ls1 ls2)
```

A continuación se muestran ejemplos de expresiones construidas con las funciones anteriores:

```
exp1 : Exp ["x", "y"]
exp1 = add (var "x") (add (cons 1) (var "y"))

exp2 : Exp ["y"]
exp2 = eLet ("x" := 10) $ add (var "x") (var "y")

exp3 : Exp []
exp3 = eLet ("y" := 5) exp4
```

También se decidió incluir otra función útil para construir expresiones. Se trata de *local*, que permite tomar una lista de declaraciones de variables y definir la sustitución de todas en una subexpresión:

```
data LocalVariables : List String -> Type where
  Nil : LocalVariables []
  (::) : VarDec l -> LocalVariables ls ->
      LocalVariables (l :: ls)

localPred : (vars : LocalVariables localVars) ->
      (innerExp : Exp fvsInner) -> {isSet : IsSet localVars} ->
      Exp (projectRightList localVars fvsInner)

local : (vars : LocalVariables localVars) ->
      (innerExp : Exp fvsInner) ->
```

```
TypeOrUnit
   (isSet localVars)
   (Exp (projectRightList localVars fvsInner))
local {localVars} {fvsInner} vars innerExp =
   mkTypeOrUnit (isSet localVars)
   (\localIsSet => localPred vars innerExp {isSet=localIsSet})
```

LocalVariables es un tipo que contiene una lista de declaraciones locales de variables, y en su tipo mantiene la lista de variables declaradas. Por ejemplo, se tiene ['x' := 10, 'y' := 4] : LocalVariables ['x', 'y'].

localPred toma la declaración de variables locales, una subexpresión, una prueba de que las variables declaradas no se repiten, y retorna una nueva expresión. Su implementación se encuentra en el anexo, pero en términos simples construye Lets secuencialmente para cada variable declarada en la lista. La prueba IsSet localVars de que las variables no son repetidas nunca es utilizada, simplemente se agrega para restringir el uso de local y no poder incluir variables repetidas (lo cual sí es posible en aplicaciones sucesivas de Let).

local utiliza el mismo truco definido para Record anteriormente para poder generar la prueba de IsSet localVars de forma automática.

Con estas definiciones se pueden construir expresiones de la siguiente manera:

```
exp1 : Exp []
exp1 = local ["x" := 10] $ cons 1

exp2 : Exp []
exp2 = local ["x" := 10, "y" := 9] $ add (var "x") (var "y")
```

### 4.3. Interpretación de una expresión

**PENDIENTE** 

## Conclusión

**PENDIENTE** 

#### 5.1. Alternativas de HList en Idris

El sistema de tipos de Idris permite definir tipos de muchas maneras, por lo cual en su momento se investigaron otras formas distintas de implementar HList. A continuacion se describen tales formas, indicando por que no se opto por cada una ellas.

#### 5.1.1. Dinámico

```
data HValue : Type where
    HVal: {A : Type} -> (x : A) -> HValue

HList : Type
HList = List HValue
```

Este tipo se asemeja al tipo Dynamic utilizado a veces en Haskell, o a Object en Java/C#. Esta HList mantiene valores de tipos arbitrarios dentro de ella, pero no proporciona ninguna informacion de ellos en su tipo. Cada valor es simplemente reconocido como HValue, y no es posible conocer su tipo u operar con el de ninguna forma. Solo es posible insertar elementos, y manipularlos en la lista (eliminarlos, reordenarlos, etc).

Este enfoque se asemeja al uso de List<Object> de Java/C# que fue usado incluso antes de que estos lenguajes tuvieran tipos parametrizables, pero sin la funcionalidad de poder realizar down-casting (castear un elemento de una superclase a una subclase).

No se utilizo este enfoque ya que al no poder obtenerse la informacion del tipo de HValue es imposible poder verificar que un record contenga campos con etiquetas y que estos no esten repetidos, al igual que es imposible poder trabajar con tal record luego de construido.

Un ejemplo de su uso es

```
[HVal (1,2), HVal "Hello", HVal 42] : HList
```

#### 5.1.2. Existenciales

```
data HList : Type where
   Nil : HList
   (::) : {A : Type} -> (x : A) -> HList -> HList
```

Este enfoque se asemeja al uso de *tipos existenciales* utilizado en Haskell. Básicamente el tipo HList se define como un tipo simple sin parámetros, pero sus constructores permiten utilizar valores de cualquier tipo. Esta definicion es muy similar a la que utiliza tipos dinámicos, y tiene las mismas desventajas. Luego de creada una HList de esta forma, no es posible obtener ninguna informacion de los tipos de los valores que contiene, por lo que es imposible poder trabajar con tales valores. Por ese motivo tampoco fue utilizado.

Un ejemplo de su uso es

```
[1,"2"] : HList
```

#### 5.1.3. Estructurado

Esta definicion es un punto medio (en terminos de poder) entre la definicion utilizada en este trabajo y las demás definiciones descritas en las secciones anteriores.

Esta HList es parametrizada sobre un constructor de tipos. Es decir, toma como parámetro una funcion que toma un tipo y construye otro tipo a partir de este. Esta definicion permite imponer una estructura en común a todos los elementos de la lista, forzando que cada uno de ellos haya sido construido con tal constructor de tipo, sin importar el tipo base utilizado. La desventaja es que no es posible conocer nada de los tipos de cada elemento sin ser por la estructura que se impone en su tipo. El tipo utilizado en este trabajo (al igual que el tipo <code>HList</code> utilizado por Idris) permite utilizar tipos arbitrarios y obtener informacion de ellos accediendo a la lista de tipos, por lo cual son más útiles.

Algunos ejemplos son los siguientes:

```
hListTuple : HList (\x => (x, x))
hListTuple = (1,1) :: ("1","2") :: Nil
hListExample : HList id
hListExample = 1 :: "1" :: (1,2) :: Nil
```

Como se ve en el último ejemplo, se puede reconstruir la definicion de HList existencial simple utilizando  $\mathtt{HList}$  id.

## Trabajo a futuro

En este trabajo se encontraron varias tareas para mejorar esta implementación de records extensibles.

- Una posible tarea a futuro es terminar de implementar las demás funciones y predicados de *HList* de Haskell que faltan implementar.
- En el diseño de la biblioteca y del caso de estudio, hubo varios predicados y funciones duplicados para List (lty, Type) y List String. Predicados como IsLeftUnion, DeleteLabelAt, y muchas otras. Una mejora es crear predicados y funciones parametrizadas por List a, con una restricción sobre este a de que contenga una etiqueta o valor String. De este modo no se repite código ni se realiza más esfuerzo en probar todos los lemas para cada predicado en particular.
- Todavia existen muchos lemas y teoremas que relacionan todos los predicados y tipos creados que podrían ser muy útiles. Es fundamental que la biblioteca contenga estos teoremas para que el programador que la use no tenga que probarlos por sí solo cada vez (lo cual va a requerir que el usuario de esta biblioteca conozca su comportamiento interno, que es necesario para probar los teoremas).
- Para algunos predicados se crearon funciones que computan su mismo resultado. Por ejemplo, para el predicado IsleftUnion se implementó la función
  leftUnion. Un buen trabajo a futuro sería implementar las funciones análogas de todos los predicados definidos, y definir todos los posibles teoremas y
  lemas interesantes entre esas funciones y los distintos predicados.
- La implementación actual funciona sobre listas ordenadas. Esto hace que Record [('A', Nat), ('B', Nat)] sea distinto de Record [('B', Nat), ('A', Nat)], cuando en realidad tal orden no es necesario para utilizar las funciones definidas para records. Un posible trabajo es parametrizar el tipo del record por un tipo abstracto. Este tipo abstracto debe funcionar como una colección de etiquetas y tipos, y debe permitir operaciones como head, index, cons, nil, entre otras. Al funcionar como tipo abstracto se podría utilizar cualquier tipo que cumpla esas propiedades, eligiendo el que se adecúe a la situación del usuario. Se podría utilizar árboles binarios, árboles Red-Black, Fingertrees, etc.

- Actualmente los records se implementaron siguiendo los pasos de *HList* de Haskell. Sin embargo, pueden existir otras implementaciones que sean mejores en otras situaciones. Una buena tarea a futuro es diseñar la biblioteca utilizando el record como tipo abstracto, tal que la implementación misma del record pueda variar sin que los usuarios de la biblioteca sean afectados.
- Como se vió en la introducción, los conceptos visitados en este trabajo aplican tanto a Agda como a Idris. Sería interesante estudiar replicar esta implementación en Agda.
- Se puede mejorar los mensajes de error en la generación automática de errores. Cuando se utiliza TypeOrUnit, si el compilador no puede generar el predicado esperado, entonces se muestra un error de unificación de TypeOrUnit con el tipo esperado. Se puede investigar una nueva forma de utilizar TypeOrUnit para mostrar mejores mensajes de error. Se debería poder asignar un mensaje de error a cada aplicación de TypeOrUnit, como por ejemplo el mensaje . El elemento a agregar ya existe" al llamar a consRec.

# **Apéndice**

## 7.1. Código fuente

PENDIENTE

## Bibliografía

- [1] .
- [2] .
- [3] .
- [4] .
- [5] .
- [6] .
- [7] .
- [8]
- [9] An overview of Hugs extensions Extensible records: Trex. 1999. URL: https: //www.haskell.org/hugs/pages/hugsman/exts.html#sect7. 2 (visitado 07-02-2017).
- [10] Edwin Brady. «Idris, a general-purpose dependently typed programming language: Design and implementation». En: Journal of Functional Programming 23 (sep. de 2013), págs. 552-593. ISSN: 1469-7653. DOI: 10.1017/ s095679681300018x.URL: http://dx.doi.org/10.1017/s095679681300018x.
- [11] Luca Cardelli y John C. Mitchell. «Operations on Records». En: Proceedings of the Fifth International Conference on Mathematical Foundations of Programming Semantics. New Orleans, Louisiana, USA: Springer-Verlag New York, Inc., 1990, págs. 22-52. ISBN: 0-387-97375-3. URL: http://dl.acm. org/citation.cfm?id=101514.101515.
- [12] Evan Czaplicki y Stephen Chong. «Asynchronous Functional Reactive Programming for GUIs». En: SIGPLAN Not. 48.6 (jun. de 2013), págs. 411-422. ISSN: 0362-1340. DOI: 10.1145/2499370.2462161. URL: http:// doi.acm.org/10.1145/2499370.2462161.
- [13] Elm Compilers as Assistants. 2015. URL: http://elm-lang.org/ blog/compilers-as-assistants (visitado 17-04-2016).
- [14] Elm Records. 2016. URL: http://elm-lang.org/docs/records (visitado 08-03-2016).
- [15] Benedict R. Gaster y Mark P. Jones. A Polymorphic Type System for Extensible Records and Variants. Inf. téc. 1996.

BIBLIOGRAFÍA 46

[16] Wolfgang Jeltsch. «Generic Record Combinators with Static Type Checking». En: Proceedings of the 12th International ACM SIGPLAN Symposium on Principles and Practice of Declarative Programming. PPDP '10. Hagenberg, Austria: ACM, 2010, págs. 143-154. ISBN: 978-1-4503-0132-9. DOI: 10.1145/1836089.1836108. URL: http://doi.acm.org/10.1145/1836089.1836108.

- [17] Mark P. Jones y Simon Peyton Jones. Lightweight Extensible Records for Haskell. 1999.
- [18] Oleg Kiselyov, Ralf Lämmel y Keean Schupke. *Hackage The HList package*. 2004. URL: https://hackage.haskell.org/package/HList(visitado 06-03-2016).
- [19] Oleg Kiselyov, Ralf Lämmel y Keean Schupke. «Strongly Typed Heterogeneous Collections». En: Proceedings of the 2004 ACM SIGPLAN Workshop on Haskell. Haskell '04. Snowbird, Utah, USA: ACM, 2004, págs. 96-107. ISBN: 1-58113-850-4. DOI: 10.1145/1017472.1017488. URL: http://doi.acm.org/10.1145/1017472.1017488.
- [20] Daan Leijen. «Extensible records with scoped labels». En: *Proceedings of the 2005 Symposium on Trends in Functional Programming (TFP'05)*. Tallin, Estonia, sep. de 2005.
- [21] Daan Leijen. First-class labels for extensible rows. Inf. téc. UU-CS-2004-51. Department of Computer Science, Universiteit Utrecht, dic. de 2004.
- [22] Ulf Norell. «Dependently Typed Programming in Agda». En: Proceedings of the 4th International Workshop on Types in Language Design and Implementation. TLDI '09. Savannah, GA, USA: ACM, 2009, págs. 1-2. ISBN: 978-1-60558-420-1. DOI: 10.1145/1481861.1481862. URL: http://doi.acm.org/10.1145/1481861.1481862.
- [23] Purescript Handling Native Effects with the Eff Monad. 2016. URL: http://www.purescript.org/learn/eff/ (visitado 17-04-2016).
- [24] Purescript by Example. 2016. URL: https://leanpub.com/purescript (visitado 31-01-2017).